

Modal quantales, involutive Quantales, Dedekind Quantales

Cameron Calk and Georg Struth

September 13, 2023

Abstract

This AFP entry provides mathematical components for modal quantales, involutive quantales and Dedekind quantales. Modal quantales are simple extensions of modal Kleene algebras useful for the verification of recursive programs. Involutive quantales appear in the study of C^* -algebras. Dedekind quantales are relatives of Tarski's relation algebras, hence relevant to program verification and beyond that to higher rewriting. We also provide components for weaker variants such as Kleene algebras with converse and modal Kleene algebras with converse.

Contents

1	Introductory Remarks	2
2	Modal Kleene algebra based on domain and range semirings	3
2.1	Modal semirings	3
2.2	Modal Kleene algebra	3
3	Kleene algebra with converse	4
3.1	Involutive Kleene algebra	4
3.2	Kleene algebra with converse	6
4	Modal Kleene algebra with converse	7
4.1	Involutive modal Kleene algebras	7
4.2	Modal semirings algebras with converse	7
4.3	Modal Kleene algebras with converse	9
5	Modal quantales	9
5.1	Simplified modal semirings and Kleene algebras	9
5.2	Domain quantales	10
5.3	Codomain quantales	14
5.4	Modal quantales	15
5.5	Antidomain and anticodomain quantales	15

6	Quantales with converse	16
6.1	Properties of unital quantales	16
6.2	Involutive quantales	17
6.3	Dedekind quantales	23
6.4	Boolean Dedekind quantales	44

1 Introductory Remarks

In this AFP entry we provide mathematical components for modal quantales, involutive quantales and Dedekind quantales. Modal quantales are simple extensions of modal Kleene algebras that can be used in the verification of recursive programs [6]. Involutive quantales appear in the study of C^* -algebras [8]. Dedekind quantales, categorifications of which are known as *modular quantaloids* [9], are relatives of Tarski’s relation algebras [11], and hence relevant to program verification as well. We also provide components for weaker variants such as Kleene algebras and modal Kleene algebras with converse.

Our main interest in these structures comes from recent applications in higher-dimensional rewriting [2, 3], where they are used in coherence proofs for rewriting systems based on computads or polygraphs. This includes proofs of coherent Church-Rosser theorems and coherent Newman’s lemmas. A more long-term programme considers the formalisation of algebraic aspects of higher rewriting with proof assistants.

Modal quantales have previously been studied in [4], where it is shown, for instance, that any category can be lifted to a modal quantale at powerset level. Such lifting results will be formalised in a companion AFP entry.

Dedekind quantales give also rise to intuitionistic modal algebras, as the results in this AFP entry show. In particular, the set of all subidentities or coreflexives of a Dedekind quantale forms a complete Heyting algebra (aka frame or locale), on which modal box and diamond operators can be defined. A paper explaining these results is in preparation [7]. A further application of Dedekind quantales lies once again in higher-dimensional rewriting [2, 3]. Any groupoid, in particular, can be lifted to a Dedekind quantale at powerset level, a result which will once again be formalised in a companion AFP entry. Our components build on extant AFP components for Kleene algebras [1], modal Kleene algebras [5] and quantales [10].

Georg Struth is grateful for an invited professorship at École polytechnique and a fellowship at the Collegium de Lyon, Institute of Advanced Study, during which most of this formalisation work has been done.

2 Modal Kleene algebra based on domain and range semirings

```
theory Modal-Kleene-Algebra-Var
  imports KAD.Domain-Semiring KAD.Range-Semiring
```

```
begin
```

```
notation domain-op (dom)
```

```
notation range-op (cod)
```

```
subclass (in domain-semiring) dioid-one-zero..
```

```
subclass (in range-semiring) dioid-one-zero
  by unfold-locales simp
```

2.1 Modal semirings

The following modal semirings are based on domain and range semirings instead of antidomain and antirange semirings, as in the AFP entry for Kleene algebra with domain.

```
class dr-modal-semiring = domain-semiring + range-semiring +
  assumes dc-compat [simp]: dom (cod x) = cod x
  and cd-compat [simp]: cod (dom x) = dom x
```

```
begin
```

```
sublocale msrdual: dr-modal-semiring (+)  $\lambda x y. y \cdot x$  1 0 cod ( $\leq$ ) ( $<$ ) dom
  by unfold-locales simp-all
```

```
lemma d-cod-fix: (dom x = x) = (x = cod x)
  by (metis local.cd-compat local.dc-compat)
```

```
lemma local-var: (x · y = 0) = (cod x · dom y = 0)
  using local.dom-weakly-local local.rdual.dom-weakly-local by force
```

```
lemma fbdia-conjugation: (fd x (dom p) · dom q = 0) = (dom p · bd x (dom q) = 0)
  by (metis local.bd-def local.cd-compat local.ddual.mult-assoc local.dom-weakly-local
    local.fd-def local.rdual.dom-weakly-local local.rdual.dsg4)
```

```
end
```

2.2 Modal Kleene algebra

```
class dr-modal-kleene-algebra = dr-modal-semiring + kleene-algebra
```

```
end
```

3 Kleene algebra with converse

```
theory Kleene-Algebra-Converse
  imports Kleene-Algebra.Kleene-Algebra
```

```
begin
```

We start from involutive dioids and Kleene algebra and then add a so-called strong Gelfand property to obtain an operation of converse that is closer to algebras of paths and relations.

3.1 Involutive Kleene algebra

```
class invol-op =
  fixes invol :: 'a ⇒ 'a (-° [101] 100)

class involutive-dioid = dioid-one-zero + invol-op +
  assumes inv-invol [simp]: (x°)° = x
  and inv-contrav [simp]: (x · y)° = y° · x°
  and inv-sup [simp]: (x + y)° = x° + y°
```

```
begin
```

```
lemma inv-zero [simp]: 0° = 0
```

```
proof -
```

```
  have 0° = (0° · 0)°
```

```
    by simp
```

```
  also have ... = 0° · (0°)°
```

```
    using local.inv-contrav by blast
```

```
  also have ... = 0° · 0
```

```
    by simp
```

```
  also have ... = 0
```

```
    by simp
```

```
  finally show ?thesis.
```

```
qed
```

```
lemma inv-one [simp]: 1° = 1
```

```
proof -
```

```
  have 1° = 1° · (1°)°
```

```
    by simp
```

```
  also have ... = (1° · 1)°
```

```
    using local.inv-contrav by presburger
```

```
  also have ... = (1°)°
```

```
    by simp
```

```
  also have ... = 1
```

```
    by simp
```

```
  finally show ?thesis.
```

```
qed
```

lemma *inv-iso*: $x \leq y \implies x^\circ \leq y^\circ$
by (*metis local.inv-sup local.less-eq-def*)

lemma *inv-adj*: $(x^\circ \leq y) = (x \leq y^\circ)$
using *inv-iso* **by** *fastforce*

end

Here is an equivalent axiomatisation from Doornbos, Backhouse and van der Woude's paper on a calculational approach to mathematical induction.

class *involutive-dioid-alt* = *dioid-one-zero* +
fixes *inv-alt* :: 'a \Rightarrow 'a
assumes *inv-alt*: $(\text{inv-alt } x \leq y) = (x \leq \text{inv-alt } y)$
and *inv-alt-contrav* [*simp*]: $\text{inv-alt } (x \cdot y) = \text{inv-alt } y \cdot \text{inv-alt } x$

begin

lemma *inv-alt-invol* [*simp*]: $\text{inv-alt } (\text{inv-alt } x) = x$

proof–

have $\text{inv-alt } (\text{inv-alt } x) \leq x$

by (*simp add: inv-alt*)

thus *?thesis*

by (*meson inv-alt order-antisym*)

qed

lemma *inv-alt-add*: $\text{inv-alt } (x + y) = \text{inv-alt } x + \text{inv-alt } y$

proof–

{fix *z*

have $(\text{inv-alt } (x + y) \leq z) = (x + y \leq \text{inv-alt } z)$

by (*simp add: inv-alt*)

also have $\dots = (x \leq \text{inv-alt } z \wedge y \leq \text{inv-alt } z)$

by *simp*

also have $\dots = (\text{inv-alt } x \leq z \wedge \text{inv-alt } y \leq z)$

by (*simp add: inv-alt*)

also have $\dots = (\text{inv-alt } x + \text{inv-alt } y \leq z)$

by *force*

finally have $(\text{inv-alt } (x + y) \leq z) = (\text{inv-alt } x + \text{inv-alt } y \leq z).$

thus *?thesis*

using *order-antisym* **by** *blast*

qed

sublocale *altinv*: *involutive-dioid* - - - - - *inv-alt*

by *unfold-locales (simp-all add: inv-alt-add)*

end

sublocale *involutive-dioid* \subseteq *altinv*: *involutive-dioid-alt* - - - - - *invol*

by *unfold-locales (simp-all add: local.inv-adj)*

```

class involutive-kleene-algebra = involutive-diod + kleene-algebra

begin

lemma inv-star:  $(x^*)^\circ = (x^\circ)^*$ 
proof (rule order.antisym)
  have  $((x^\circ)^*)^\circ = (1 + (x^\circ)^* \cdot x^\circ)^\circ$ 
    by simp
  also have  $\dots = 1 + (x^\circ)^\circ \cdot ((x^\circ)^*)^\circ$ 
    using local.inv-contrav local.inv-one local.inv-sup by presburger
  finally have  $1 + x \cdot ((x^\circ)^*)^\circ \leq ((x^\circ)^*)^\circ$ 
    by simp
  hence  $x^* \leq ((x^\circ)^*)^\circ$ 
    using local.star-inductl by force
  thus  $(x^*)^\circ \leq (x^\circ)^*$ 
    by (simp add: local.inv-adj)
next
  have  $(x^*)^\circ = (1 + x^* \cdot x)^\circ$ 
    by simp
  also have  $\dots = 1 + x^\circ \cdot (x^*)^\circ$ 
    using local.inv-contrav local.inv-one local.inv-sup by presburger
  finally have  $1 + x^\circ \cdot (x^*)^\circ \leq (x^*)^\circ$ 
    by simp
  thus  $(x^\circ)^* \leq (x^*)^\circ$ 
    using local.star-inductl by force
qed

end

```

3.2 Kleene algebra with converse

The name "strong Gelfand property" has been borrowed from Palmigiano and Re.

```

class dioid-converse = involutive-diod +
  assumes strong-gelfand:  $x \leq x \cdot x^\circ \cdot x$ 

lemma (in dioid-converse) subid-conv:  $x \leq 1 \implies x^\circ = x$ 
proof (rule order.antisym)
  assume h:  $x \leq 1$ 
  have  $x \leq x \cdot x^\circ \cdot x$ 
    by (simp add: local.strong-gelfand)
  also have  $\dots \leq 1 \cdot x^\circ \cdot 1$ 
    using h local.mult-isol-var by blast
  also have  $\dots = x^\circ$ 
    by simp
  finally show  $x \leq x^\circ$ 
    by simp
  thus  $x^\circ \leq x$ 
    by (simp add: local.inv-adj)

```

qed

class *kleene-algebra-converse* = *involutive-kleene-algebra* + *diod-converse*

end

4 Modal Kleene algebra with converse

theory *Modal-Kleene-Algebra-Converse*

imports *Modal-Kleene-Algebra-Var Kleene-Algebra-Converse*

begin

Here we mainly study the interaction of converse with domain and codomain.

4.1 Involutive modal Kleene algebras

class *involutive-domain-semiring* = *domain-semiring* + *involutive-diod*

begin

notation *domain-op* (*dom*)

lemma *strong-conv-conv*: $\text{dom } x \leq x \cdot x^\circ \implies x \leq x \cdot x^\circ \cdot x$

proof –

assume *h*: $\text{dom } x \leq x \cdot x^\circ$

have $x = \text{dom } x \cdot x$

by *simp*

also have $\dots \leq x \cdot x^\circ \cdot x$

using *h local.mult-isor* by *presburger*

finally show *?thesis*.

qed

end

class *involutive-dr-modal-semiring* = *dr-modal-semiring* + *involutive-diod*

class *involutive-dr-modal-kleene-algebra* = *involutive-dr-modal-semiring* + *kleene-algebra*

4.2 Modal semirings algebras with converse

class *dr-modal-semiring-converse* = *dr-modal-semiring* + *diod-converse*

begin

lemma *d-conv* [*simp*]: $(\text{dom } x)^\circ = \text{dom } x$

proof –

have $\text{dom } x \leq \text{dom } x \cdot (\text{dom } x)^\circ \cdot \text{dom } x$

by (*simp add: local.strong-gelfand*)

also have $\dots \leq 1 \cdot (\text{dom } x)^\circ \cdot 1$
by (*simp add: local.subid-conv*)
finally have $a: \text{dom } x \leq (\text{dom } x)^\circ$
by *simp*
hence $(\text{dom } x)^\circ \leq \text{dom } x$
by (*simp add: local.inv-adj*)
thus *?thesis*
using a **by** *auto*
qed

lemma *cod-conv*: $(\text{cod } x)^\circ = \text{cod } x$
by (*metis d-conv local.dc-compat*)

lemma *d-conv-cod* [*simp*]: $\text{dom } (x^\circ) = \text{cod } x$
proof –

have $\text{dom } (x^\circ) = \text{dom } ((x \cdot \text{cod } x)^\circ)$
by *simp*
also have $\dots = \text{dom } ((\text{cod } x)^\circ \cdot x^\circ)$
using *local.inv-contrav* **by** *presburger*
also have $\dots = \text{dom } (\text{cod } x \cdot x^\circ)$
by (*simp add: cod-conv*)
also have $\dots = \text{dom } (\text{dom } (\text{cod } x) \cdot x^\circ)$
by *simp*
also have $\dots = \text{dom } (\text{cod } x) \cdot \text{dom } (x^\circ)$
using *local.dsg3* **by** *blast*
also have $\dots = \text{cod } x \cdot \text{dom } (x^\circ)$
by *simp*
also have $\dots = \text{cod } x \cdot \text{cod } (\text{dom } (x^\circ))$
by *simp*
also have $\dots = \text{cod } (x \cdot \text{cod } (\text{dom } (x^\circ)))$
using *local.rdual.dsg3* **by** *presburger*
also have $\dots = \text{cod } (x \cdot \text{dom } (x^\circ))$
by *simp*
also have $\dots = \text{cod } ((x^\circ)^\circ \cdot (\text{dom } (x^\circ))^\circ)$
by *simp*
also have $\dots = \text{cod } ((\text{dom } (x^\circ) \cdot x^\circ)^\circ)$
using *local.inv-contrav* **by** *presburger*
also have $\dots = \text{cod } ((x^\circ)^\circ)$
by *simp*
also have $\dots = \text{cod } x$
by *simp*
finally show *?thesis*.
qed

lemma *cod-conv-d*: $\text{cod } (x^\circ) = \text{dom } x$
by (*metis d-conv-cod local.inv-invol*)

lemma $\text{dom } y = y \implies \text{fd } (x^\circ) y = \text{bd } x y$
proof –


```

assume  $h: \text{dom } y = y$ 
have  $\text{fd } (x^\circ) y = \text{dom } (x^\circ \cdot \text{dom } y)$ 
  by (simp add: local.fd-def)
also have  $\dots = \text{dom } ((\text{dom } y \cdot x)^\circ)$ 
  by simp
also have  $\dots = \text{cod } (\text{dom } y \cdot x)$ 
  using d-conv-cod by blast
also have  $\dots = \text{bd } x y$ 
  by (simp add: h local.bd-def)
finally show ?thesis.
qed

```

```

lemma  $\text{dom } y = y \implies \text{bd } (x^\circ) y = \text{fd } x y$ 
  by (metis cod-conv-d d-conv local.bd-def local.fd-def local.inv-contrav)

```

end

4.3 Modal Kleene algebras with converse

```

class dr-modal-kleene-algebra-converse = dr-modal-semiring-converse + kleene-algebra

```

```

class dr-modal-semiring-strong-converse = involutive-dr-modal-semiring +
  assumes weak-dom-def:  $\text{dom } x \leq x \cdot x^\circ$ 
  and weak-cod-def:  $\text{cod } x \leq x^\circ \cdot x$ 

```

```

subclass (in dr-modal-semiring-strong-converse) dr-modal-semiring-converse
  by unfold-locales (metis local.ddual.mult-isol-var local.dsg1 local.eq-refl local.weak-dom-def)

```

```

class dr-modal-kleene-algebra-strong-converse = dr-modal-semiring-strong-converse
+ kleene-algebra

```

end

5 Modal quantales

```

theory Modal-Quantale

```

```

  imports Quantales.Quantale-Star Modal-Kleene-Algebra-Var KAD.Modal-Kleene-Algebra

```

begin

5.1 Simplified modal semirings and Kleene algebras

The previous formalisation of modal Kleene algebra in the AFP adds two compatibility axioms between domain and codomain when combining an antidomain semiring with an antirange semiring. But these are unnecessary. They are derivable from the other axioms. Thus I provide a simpler axiomatisation that should eventually replace the one in the AFP.

```

class modal-semiring-simp = antidomain-semiring + antirange-semiring

```

```

lemma (in modal-semiring-simp) dr-compat [simp]:  $d (r x) = r x$ 
proof –
  have  $a: ar x \cdot d (r x) = 0$ 
    using local.ads-d-def local.ars-r-def local.dpdz.dom-weakly-local by auto
  have  $rx \cdot d (r x) \cdot ar x \leq rx \cdot ar x$ 
    by (simp add: local.a-subid-aux2 local.ads-d-def local.mult-isor)
  hence  $b: rx \cdot d (r x) \cdot ar x = 0$ 
    by (simp add: local.ardual.am2 local.ars-r-def local.join.bot-unique)
  have  $d (r x) = (ar x + r x) \cdot d (r x)$ 
    using local.add-comm local.ardual.ans3 local.ars-r-def local.mult-1-left by pres-
burger
  also have  $\dots = ar x \cdot d (r x) + r x \cdot d (r x)$ 
    by simp
  also have  $\dots = r x \cdot d (r x)$ 
    by (simp add: a)
  also have  $\dots = r x \cdot d (r x) \cdot (ar x + r x)$ 
    using local.add-comm local.ardual.ans3 local.ars-r-def by auto
  also have  $\dots = r x \cdot d (r x) \cdot ar x + r x \cdot d (r x) \cdot r x$ 
    by simp
  also have  $\dots = r x \cdot d (r x) \cdot r x$ 
    using b by auto
  also have  $\dots = r x$ 
    by (metis local.ads-d-def local.am3 local.ardual.a-mult-idem local.ars-r-def lo-
cal.ds.ddual.mult-assoc)
  finally show ?thesis
    by simp
qed

```

```

lemma (in modal-semiring-simp) rd-compat [simp]:  $r (d x) = d x$ 
  by (smt (verit) local.a-mult-idem local.ads-d-def local.am2 local.ardual.dpdz.dom-weakly-local
local.ars-r-def local.dr-compat local.kat-3-equiv')

```

```

subclass (in modal-semiring-simp) modal-semiring
  apply unfold-locales by simp-all

```

```

class modal-kleene-algebra-simp = modal-semiring-simp + kleene-algebra

```

```

subclass (in modal-kleene-algebra-simp) modal-kleene-algebra..

```

5.2 Domain quantales

```

class domain-quantale = unital-quantale + domain-op +
  assumes dom-absorb:  $x \leq dom x \cdot x$ 
  and dom-local:  $dom (x \cdot dom y) = dom (x \cdot y)$ 
  and dom-add:  $dom (x \sqcup y) = dom x \sqcup dom y$ 
  and dom-subid:  $dom x \leq 1$ 
  and dom-zero [simp]:  $dom \perp = \perp$ 

```

The definition is that of a domain semiring. I cannot extend the quantale

class with respect to domain semirings because of different operations are used for addition/sup. The following sublocale statement brings all those properties into scope.

```
sublocale domain-quantale  $\subseteq$  dqmsr: domain-semiring ( $\sqcup$ ) ( $\cdot$ )  $1 \perp$  dom ( $\leq$ ) ( $<$ )
  by unfold-locales (simp-all add: dom-add dom-local dom-absorb sup.absorb2 dom-subid)
```

```
sublocale domain-quantale  $\subseteq$  dqmka: domain-kleene-algebra ( $\sqcup$ ) ( $\cdot$ )  $1 \perp$  dom ( $\leq$ )
( $<$ ) qstar..
```

```
typedef (overloaded) 'a d-element = {x :: 'a :: domain-quantale. dom x = x}
  using dqmsr.dom-one by blast
```

```
setup-lifting type-definition-d-element
```

```
instantiation d-element :: (domain-quantale) bounded-lattice
```

```
begin
```

```
lift-definition less-eq-d-element :: 'a d-element  $\Rightarrow$  'a d-element  $\Rightarrow$  bool is ( $\leq$ ) .
```

```
lift-definition less-d-element :: 'a d-element  $\Rightarrow$  'a d-element  $\Rightarrow$  bool is ( $<$ ) .
```

```
lift-definition bot-d-element :: 'a d-element is  $\perp$ 
  by simp
```

```
lift-definition top-d-element :: 'a d-element is  $1$ 
  by simp
```

```
lift-definition inf-d-element :: 'a d-element  $\Rightarrow$  'a d-element  $\Rightarrow$  'a d-element is ( $\cdot$ )
  by (metis dqmsr.dom-mult-closed)
```

```
lift-definition sup-d-element :: 'a d-element  $\Rightarrow$  'a d-element  $\Rightarrow$  'a d-element is
( $\sqcup$ )
  by simp
```

```
instance
```

```
  apply (standard; transfer)
    apply (simp-all add: less-le-not-le)
      apply (metis dqmsr.dom-subid-aux2'')
      apply (metis dqmsr.dom-subid-aux1'')
      apply (metis dqmsr.dom-glb-eq)
      by (metis dqmsr.dom-subid)
```

```
end
```

```
instance d-element :: (domain-quantale) distrib-lattice
  by (standard, transfer, metis dqmsr.dom-distrib)
```

```
context domain-quantale
```

begin

lemma *dom-top* [*simp*]: $\text{dom } \top = 1$

proof –

have $1 \leq \top$

by *simp*

hence $\text{dom } 1 \leq \text{dom } \top$

using *local.dqmsr.d-iso* **by** *blast*

thus *?thesis*

by (*simp add: local.dual-order.antisym*)

qed

lemma *dom-top2*: $x \cdot \top \leq \text{dom } x \cdot \top$

proof –

have $x \cdot \top = \text{dom } x \cdot x \cdot \top$

by *simp*

also have $\dots \leq \text{dom } x \cdot \top \cdot \top$

using *local.dqmsr.d-restrict-iff-1 local.top-greatest local.top-times-top mult-assoc*

by *presburger*

finally show *?thesis*

by (*simp add: local.mult.semigroup-axioms semigroup.assoc*)

qed

lemma *weak-twisted*: $x \cdot \text{dom } y \leq \text{dom } (x \cdot y) \cdot x$

using *local.dqmsr.fd-def local.dqmsr.fdemodalisation2 local.eq-refl* **by** *blast*

lemma *dom-meet*: $\text{dom } x \cdot \text{dom } y = \text{dom } x \sqcap \text{dom } y$

apply (*rule order.antisym*)

apply (*simp add: local.dqmsr.dom-subid-aux2 local.dqmsr.dom-subid-aux2''*)

by (*smt (z3) local.dom-absorb local.dqmsr.dom-iso local.dqmsr.dom-subid-aux2 local.dqmsr.dsg3 local.dqmsr.dsg4 local.dual-order.antisym local.inf.cobounded1 local.inf.cobounded2 local.psrpq.mult-isol-var*)

lemma *dom-meet-pres*: $\text{dom } (\text{dom } x \sqcap \text{dom } y) = \text{dom } x \sqcap \text{dom } y$

using *dom-meet local.dqmsr.dom-mult-closed* **by** *presburger*

lemma *dom-meet-distl*: $\text{dom } x \cdot (y \sqcap z) = (\text{dom } x \cdot y) \sqcap (\text{dom } x \cdot z)$

proof –

have $a: \text{dom } x \cdot (y \sqcap z) \leq (\text{dom } x \cdot y) \sqcap (\text{dom } x \cdot z)$

using *local.mult-isol* **by** *force*

have $(\text{dom } x \cdot y) \sqcap (\text{dom } x \cdot z) = \text{dom } ((\text{dom } x \cdot y) \sqcap (\text{dom } x \cdot z)) \cdot ((\text{dom } x \cdot y) \sqcap (\text{dom } x \cdot z))$

by *simp*

also have $\dots \leq \text{dom } ((\text{dom } x \cdot y)) \cdot ((\text{dom } x \cdot y) \sqcap (\text{dom } x \cdot z))$

using *calculation local.dqmsr.dom-iso local.dqmsr.dom-llp2 local.inf.cobounded1*

by *presburger*

also have $\dots \leq \text{dom } x \cdot ((\text{dom } x \cdot y) \sqcap (\text{dom } x \cdot z))$

by (*metis local.dqmsr.domain-1'' local.dqmsr.domain-invol local.mult-isor*)

also have $\dots \leq \text{dom } x \cdot (y \sqcap z)$

by (*meson local.dqmsr.dom-subid-aux2 local.inf-mono local.order-refl local.psrpq.mult-isol-var*)
finally show *?thesis*
using *a local.dual-order.antisym* **by** *blast*
qed

lemma *dom-meet-approx*: $\text{dom} ((\text{dom } x \cdot y) \sqcap (\text{dom } x \cdot z)) \leq \text{dom } x$
by (*metis dom-meet-distl local.dqmsr.domain-1'' local.dqmsr.domain-invol*)

lemma *dom-inf-pres-aux*: $Y \neq \{\}$ $\implies \text{dom} (\prod y \in Y. \text{dom } x \cdot y) \leq \text{dom } x$
proof –

assume $Y \neq \{\}$
have $\forall z \in Y. \text{dom} (\prod y \in Y. \text{dom } x \cdot y) \leq \text{dom} (\text{dom } x \cdot z)$
by (*meson local.INF-lower local.dqmsr.dom-iso*)
hence $\forall z \in Y. \text{dom} (\prod y \in Y. \text{dom } x \cdot y) \leq \text{dom } x \cdot \text{dom } z$
by *fastforce*
hence $\forall z \in Y. \text{dom} (\prod y \in Y. \text{dom } x \cdot y) \leq \text{dom } x$
using *dom-meet* **by** *fastforce*
thus $\text{dom} (\prod y \in Y. \text{dom } x \cdot y) \leq \text{dom } x$
using $\langle Y \neq \{\} \rangle$ **by** *blast*

qed

lemma *dom-inf-pres-aux2*: $(\prod y \in Y. \text{dom } x \cdot y) \leq \prod Y$
by (*simp add: local.INF-lower2 local.dqmsr.dom-subid-aux2 local.le-Inf-iff*)

lemma *dom-inf-pres*: $Y \neq \{\}$ $\implies \text{dom } x \cdot (\prod Y) = (\prod y \in Y. \text{dom } x \cdot y)$
proof –

assume *hyp*: $Y \neq \{\}$
have $a: \text{dom } x \cdot (\prod Y) \leq (\prod y \in Y. \text{dom } x \cdot y)$
by (*simp add: Setcompr-eq-image local.Inf-subdistl*)
have $(\prod y \in Y. \text{dom } x \cdot y) = \text{dom} (\prod y \in Y. \text{dom } x \cdot y) \cdot (\prod y \in Y. \text{dom } x \cdot y)$
by *simp*
also have $\dots \leq \text{dom } x \cdot (\prod y \in Y. \text{dom } x \cdot y)$
using *dom-inf-pres-aux hyp local.mult-isol* **by** *blast*
also have $\dots \leq \text{dom } x \cdot \prod Y$
by (*simp add: dom-inf-pres-aux2 local.psrpq.mult-isol-var*)
finally show *?thesis*
using *a order.antisym* **by** *blast*

qed

lemma *dom* $(\prod X) \leq \prod (\text{dom } \cdot X)$
by (*simp add: local.INF-greatest local.Inf-lower local.dqmsr.dom-iso*)

The domain operation need not preserve arbitrary sups, though this property holds, for instance, in quantales of binary relations. I do not aim at a stronger axiomatisation in this theory.

lemma *dom-top-pres*: $(x \leq \text{dom } y \cdot x) = (x \leq \text{dom } y \cdot \top)$
apply *standard*
apply (*meson local.dqmsr.ddual.mult-isol-var local.dual-order.eq-iff local.dual-order.trans local.top-greatest*)

using *local.dqmsr.dom-iso local.dqmsr.dom-llp* **by** *fastforce*

lemma *dom-lla-var*: $(\text{dom } x \leq \text{dom } y) = (x \leq \text{dom } y \cdot \top)$
using *dom-top-pres local.dqmsr.dom-llp* **by** *force*

lemma *dom (1 \sqcap x) = 1 \sqcap x \implies x \leq 1 \implies dom x = x*
using *local.inf-absorb2* **by** *force*

lemma *dom-meet-sub*: $\text{dom } (x \sqcap y) \leq \text{dom } x \sqcap \text{dom } y$
by (*simp add: local.dqmsr.d-iso*)

lemma *dom-dist1*: $\text{dom } x \sqcup (\text{dom } y \sqcap \text{dom } z) = (\text{dom } x \sqcup \text{dom } y) \sqcap (\text{dom } x \sqcup \text{dom } z)$
by (*metis dom-meet local.dom-add local.dqmsr.dom-distrib*)

lemma *dom-dist2*: $\text{dom } x \sqcap (\text{dom } y \sqcup \text{dom } z) = (\text{dom } x \sqcap \text{dom } y) \sqcup (\text{dom } x \sqcap \text{dom } z)$
by (*metis dom-meet local.dom-add local.sup-distl*)

abbreviation *fd'* \equiv *dqmsr.fd*

definition *bb x y* = $\bigsqcup \{ \text{dom } z \mid z. \text{fd}' x z \leq \text{dom } y \}$

lemma *fd'-bb-galois-aux*: $\text{fd}' x (\text{dom } p) \leq \text{dom } q \implies \text{dom } p \leq \text{bb } x (\text{dom } q)$
by (*simp add: bb-def local.SUP-upper setcompr-eq-image*)

lemma *dom-iso-var*: $(\bigsqcup x \in X. \text{dom } x) \leq \text{dom } (\bigsqcup x \in X. \text{dom } x)$
by (*meson local.SUP-le-iff local.dom-subid local.dqmsr.domain-subid*)

lemma *dom-iso-var2*: $(\bigsqcup x \in X. \text{dom } x) \leq \text{dom } (\bigsqcup x \in X. x)$
by (*simp add: local.SUP-le-iff local.Sup-upper local.dqmsr.dom-iso*)

end

5.3 Codomain quantales

class *codomain-quantale* = *unital-quantale + range-op +*
assumes *cod-absorb*: $x \leq x \cdot \text{cod } x$
and *cod-local*: $\text{cod } (\text{cod } x \cdot y) = \text{cod } (x \cdot y)$
and *cod-add*: $\text{cod } (x \sqcup y) = \text{cod } x \sqcup \text{cod } y$
and *cod-subid*: $\text{cod } x \leq 1$
and *cod-zero*: $\text{cod } \perp = \perp$

sublocale *codomain-quantale* \subseteq *coddual: domain-quantale range-op - $\lambda x y. y \cdot x$ -*

by *unfold-locales (auto simp: mult-assoc cod-subid cod-zero cod-add cod-local cod-absorb Sup-distr Sup-distl)*

abbreviation (**in** *codomain-quantale*) *bd'* \equiv *coddual.fd'*

definition (in *codomain-quantale*) $fb\ x\ y = \sqcup \{cod\ z \mid z.\ bd'\ x\ z \leq cod\ y\}$

lemma (in *codomain-quantale*) *bd'-fb-galois-aux*: $bd'\ x\ (cod\ p) \leq cod\ q \implies cod\ p \leq fb\ x\ (cod\ q)$
using *local.coddual.bb-def local.coddual.fd'-bb-galois-aux local.fb-def* **by** *auto*

5.4 Modal quantales

class *dc-modal-quantale* = *domain-quantale* + *codomain-quantale* +
assumes *dc-compact [simp]*: $dom\ (cod\ x) = cod\ x$
and *cd-compact [simp]*: $cod\ (dom\ x) = dom\ x$

sublocale *dc-modal-quantale* \subseteq *mqs: dr-modal-kleene-algebra* (\sqcup) (\cdot) $1 \perp (\leq) (<)$
qstar dom cod
by *unfold-locales simp-all*

sublocale *dc-modal-quantale* \subseteq *mqdual: dc-modal-quantale - \lambda x y. y \cdot x - - - - -*
- - dom cod
by *unfold-locales simp-all*

lemma (in *dc-modal-quantale*) $x \cdot \top = dom\ x \cdot \top$

oops

lemma (in *dc-modal-quantale*) $\top \cdot x = \top \cdot cod\ x$

oops

5.5 Antidomain and anticodomain quantales

notation *antidomain-op* (*adom*)

class *antidomain-quantale* = *unital-quantale* + *antidomain-op* +
assumes *as1 [simp]*: $adom\ x \cdot x = \perp$
and *as2 [simp]*: $adom\ (x \cdot y) \leq adom\ (x \cdot adom\ (adom\ y))$
and *as3 [simp]*: $adom\ (adom\ x) \sqcup adom\ x = 1$

definition (in *antidomain-quantale*) $ddom = adom \circ adom$

sublocale *antidomain-quantale* \subseteq *adqmsr: antidomain-semiring adom* (\sqcup) (\cdot) $1 \perp$
 $(\leq) (<)$
by *unfold-locales (simp-all add: local.sup.absorb2)*

sublocale *antidomain-quantale* \subseteq *adqmka: antidomain-kleene-algebra adom* (\sqcup) (\cdot)
 $1 \perp (\leq) (<) qstar..$

sublocale *antidomain-quantale* \subseteq *addq: domain-quantale ddom*
by *unfold-locales (simp-all add: ddom-def local.adqmsr.ans-d-def)*

notation *antirange-op* (*acod*)

class *anticodomain-quantale* = *unital-quantale* + *antirange-op* +
 assumes *ars1* [*simp*]: $x \cdot \text{acod } x = \perp$
 and *ars2* [*simp*]: $\text{acod } (x \cdot y) \leq \text{acod } (\text{acod } (\text{acod } x) \cdot y)$
 and *ars3* [*simp*]: $\text{acod } (\text{acod } x) \sqcup \text{acod } x = 1$

sublocale *anticodomain-quantale* \subseteq *acoddual*: *antidomain-quantale* *acod* - $\lambda x y. y \cdot x$ - - - - -
 by *unfold-locales* (*auto simp: mult-assoc Sup-distl Sup-distr*)

definition (**in** *anticodomain-quantale*) *ccod* = *acod* \circ *acod*

sublocale *anticodomain-quantale* \subseteq *acdqmsr*: *antirange-semiring* (\sqcup) (\cdot) $1 \perp \text{acod}$
(\leq) ($<$)..

sublocale *anticodomain-quantale* \subseteq *acdqmka*: *antirange-kleene-algebra* (\sqcup) (\cdot) $1 \perp$
(\leq) ($<$) *qstar acod*..

sublocale *anticodomain-quantale* \subseteq *acddq*: *codomain-quantale* - - - - - λ
x. acod (*acod* *x*)
 by *unfold-locales* (*simp-all add: local.acoddual.adqmsr.ans-d-def*)

class *modal-quantale* = *antidomain-quantale* + *anticodomain-quantale*

sublocale *modal-quantale* \subseteq *mmqs*: *modal-kleene-algebra-simp* (\sqcup) (\cdot) $1 \perp (\leq) (<)$
qstar adom acod..

sublocale *modal-quantale* \subseteq *mmqdual*: *modal-quantale* - $\lambda x y. y \cdot x$ - - - - -
adom acod
 by *unfold-locales simp-all*

end

6 Quantales with converse

theory *Quantale-Converse*
 imports *Modal-Quantale Modal-Kleene-Algebra-Converse*

begin

6.1 Properties of unital quantales

These properties should eventually added to the quantales AFP entry.

lemma (**in** *quantale*) *bres-bot-top* [*simp*]: $\perp \rightarrow \top = \top$
 by (*simp add: local.bres-galois-imp local.order.antisym*)

lemma (**in** *quantale*) *fres-top-bot* [*simp*]: $\top \leftarrow \perp = \top$

by (meson local.fres-galois local.order-antisym local.top-greatest)

lemma (in unital-quantale) bres-top-top2 [simp]: $(x \rightarrow y \cdot \top) \cdot \top = x \rightarrow y \cdot \top$

proof –

have $(x \rightarrow y \cdot \top) \cdot \top \leq x \rightarrow y \cdot \top \cdot \top$

by (simp add: local.bres-interchange)

hence $(x \rightarrow y \cdot \top) \cdot \top \leq x \rightarrow y \cdot \top$

by (simp add: mult-assoc)

thus ?thesis

by (metis local.mult-1-right local.order-eq-iff local.psrpq.subdistl local.sup-top-right)

qed

lemma (in unital-quantale) fres-top-top2 [simp]: $\top \cdot (\top \cdot y \leftarrow x) = \top \cdot y \leftarrow x$

by (metis local.dual-order.antisym local.fres-interchange local.le-top local.top-greatest mult-assoc)

lemma (in unital-quantale) bres-top-bot [simp]: $\top \rightarrow \perp = \perp$

by (metis local.bot-least local.bres-canc1 local.le-top local.order.antisym)

lemma (in unital-quantale) fres-bot-top [simp]: $\perp \leftarrow \top = \perp$

by (metis local.bot-unique local.fres-canc1 local.top-le local.uqlka.independence2 local.uwqlka.star-ext)

lemma (in unital-quantale) top-bot-iff: $(x \cdot \top = \perp) = (x = \perp)$

by (metis local.fres-bot-top local.fres-canc2 local.le-bot local.mult-botl)

6.2 Involutive quantales

The following axioms for involutive quantales are standard.

class involutive-quantale = unital-quantale + invol-op +

assumes inv-invol [simp]: $(x^\circ)^\circ = x$

and inv-contrav: $(x \cdot y)^\circ = y^\circ \cdot x^\circ$

and inv-sup [simp]: $(\bigsqcup X)^\circ = (\bigsqcup x \in X. x^\circ)$

context involutive-quantale

begin

lemma inv-binsup [simp]: $(x \sqcup y)^\circ = x^\circ \sqcup y^\circ$

proof –

have $(x \sqcup y)^\circ = (\bigsqcup z \in \{x, y\}. z^\circ)$

using local.inv-sup local.sup-Sup by presburger

also have $\dots = (\bigsqcup z \in \{x^\circ, y^\circ\}. z)$

by simp

also have $\dots = x^\circ \sqcup y^\circ$

by fastforce

finally show ?thesis.

qed

lemma inv-iso: $x \leq y \implies x^\circ \leq y^\circ$

by (*metis inv-binsup local.sup.absorb-iff1*)

lemma *inv-galois*: $(x^\circ \leq y) = (x \leq y^\circ)$
 using *local.inv-iso* by *fastforce*

lemma *bres-fres-conv*: $(y^\circ \leftarrow x^\circ)^\circ = x \rightarrow y$
proof –

have $(y^\circ \leftarrow x^\circ)^\circ = (\bigsqcup \{z. z \cdot x^\circ \leq y^\circ\})^\circ$
 by (*simp add: local.fres-def*)
 also have $\dots = \bigsqcup \{z^\circ \mid z. z \cdot x^\circ \leq y^\circ\}$
 by (*simp add: image-Collect*)
 also have $\dots = \bigsqcup \{z. z^\circ \cdot x^\circ \leq y^\circ\}$
 by (*metis local.inv-invol*)
 also have $\dots = \bigsqcup \{z. (x \cdot z)^\circ \leq y^\circ\}$
 by (*simp add: local.inv-contrav*)
 also have $\dots = \bigsqcup \{z. x \cdot z \leq y\}$
 by (*metis local.inv-invol local.inv-iso*)
 also have $\dots = x \rightarrow y$
 by (*simp add: local.bres-def*)
finally show *?thesis*.

qed

lemma *fres-bres-conv*: $(y^\circ \rightarrow x^\circ)^\circ = x \leftarrow y$
 by (*metis bres-fres-conv local.inv-invol*)

sublocale *invqka*: *involutive-kleene-algebra* (\sqcup) (\cdot) $1 \perp (\leq) (<)$ *qstar invol*
 by *unfold-locales (simp-all add: local.inv-contrav)*

lemma *inv-binf [simp]*: $(x \sqcap y)^\circ = x^\circ \sqcap y^\circ$
proof –

{**fix** z
 have $(z \leq (x \sqcap y)^\circ) = (z^\circ \leq x \sqcap y)$
 using *invqka.inv-adj* by *blast*
 also have $\dots = (z^\circ \leq x \wedge z^\circ \leq y)$
 by *simp*
 also have $\dots = (z \leq x^\circ \wedge z \leq y^\circ)$
 by (*simp add: invqka.inv-adj*)
 also have $\dots = (z \leq x^\circ \sqcap y^\circ)$
 by *simp*
finally have $(z \leq (x \sqcap y)^\circ) = (z \leq x^\circ \sqcap y^\circ).$
thus *?thesis*
 using *local.dual-order.antisym* by *blast*

qed

lemma *inv-inf [simp]*: $(\prod X)^\circ = (\prod x \in X. x^\circ)$
 by (*metis invqka.inv-adj local.INF-eqI local.Inf-greatest local.Inf-lower local.inv-invol*)

lemma *inv-top [simp]*: $\top^\circ = \top$
proof –

have $a: \top^\circ \leq \top$
 by *simp*
hence $(\top^\circ)^\circ \leq \top^\circ$
 using *local.inv-iso* by *blast*
hence $\top \leq \top^\circ$
 by *simp*
thus *?thesis*
 by (*simp add: local.top-le*)
qed

lemma *inv-qstar-aux* [*simp*]: $(x \hat{\ } i)^\circ = (x^\circ) \hat{\ } i$
 by (*induct i, simp-all add: local.power-commutes*)

lemma *inv-conjugate*: $(x^\circ \sqcap y = \perp) = (x \sqcap y^\circ = \perp)$
 using *inv-binf invqka.inv-zero* by *fastforce*

We define domain and codomain as in relation algebra and compare with the domain and codomain axioms above.

definition *do* :: $'a \Rightarrow 'a$ **where**
 $do\ x = 1 \sqcap (x \cdot x^\circ)$

definition *cd* :: $'a \Rightarrow 'a$ **where**
 $cd\ x = 1 \sqcap (x^\circ \cdot x)$

lemma *do-inv*: $do\ (x^\circ) = cd\ x$

proof–

have $do\ (x^\circ) = 1 \sqcap (x^\circ \cdot (x^\circ)^\circ)$

by (*simp add: do-def*)

also have $\dots = 1 \sqcap (x^\circ \cdot x)$

by *simp*

also have $\dots = cd\ x$

by (*simp add: cd-def*)

finally show *?thesis*.

qed

lemma *cd-inv*: $cd\ (x^\circ) = do\ x$
 by (*simp add: cd-def do-def*)

lemma *do-le-top*: $do\ x \leq 1 \sqcap (x \cdot \top)$
 by (*simp add: do-def local.inf.coboundedI2 local.mult-isol*)

lemma *do-subid*: $do\ x \leq 1$
 by (*simp add: do-def*)

lemma *cd-subid*: $cd\ x \leq 1$
 by (*simp add: cd-def*)

lemma *do-bot* [*simp*]: $do\ \perp = \perp$
 by (*simp add: do-def*)

lemma *cd-bot* [*simp*]: $cd \perp = \perp$
by (*simp add: cd-def*)

lemma *do-iso*: $x \leq y \implies do x \leq do y$
by (*simp add: do-def local.inf.coboundedI2 local.inv-iso local.psrpq.mult-isol-var*)

lemma *cd-iso*: $x \leq y \implies cd x \leq cd y$
using *cd-def local.eq-refl local.inf-mono local.inv-iso local.psrpq.mult-isol-var* **by** *presburger*

lemma *do-subdist*: $do x \sqcup do y \leq do (x \sqcup y)$
proof –
have $do x \leq do (x \sqcup y)$ **and** $do y \leq do (x \sqcup y)$
by (*simp-all add: do-iso*)
thus *?thesis*
by *simp*
qed

lemma *cd-subdist*: $cd x \sqcup cd y \leq cd (x \sqcup y)$
by (*simp add: cd-iso*)

lemma *inv-do* [*simp*]: $(do x)^\circ = do x$
by (*simp add: do-def*)

lemma *inv-cd* [*simp*]: $(cd x)^\circ = cd x$
by (*metis do-inv inv-do*)

lemma *dedekind-modular*:
assumes $(x \cdot y) \sqcap z \leq (x \sqcap (z \cdot y^\circ)) \cdot (y \sqcap (x^\circ \cdot z))$
shows $(x \cdot y) \sqcap z \leq (x \sqcap (z \cdot y^\circ)) \cdot y$
using *assms local.inf.cobounded1 local.mult-isol local.order-trans* **by** *blast*

lemma *modular-eq1*:
assumes $\forall x y z w. (y \sqcap (z \cdot x^\circ) \leq w \implies (y \cdot x) \sqcap z \leq w \cdot x)$
shows $\forall x y z. (x \cdot y) \sqcap z \leq (x \sqcap (z \cdot y^\circ)) \cdot y$
using *assms* **by** *blast*

lemma $do x \cdot do y = do x \sqcap do y$
oops

lemma $p \leq 1 \implies q \leq 1 \implies p \cdot q = p \sqcap q$
oops

end

sublocale *ab-unital-quantale* \subseteq *cig: involutive-quantale id* - - - - -
by *unfold-locales (simp-all add: mult-commute)*

class *distributive-involutive-quantale* = *involutive-quantale* + *distrib-unital-quantale*

class *boolean-involutive-quantale* = *involutive-quantale* + *bool-unital-quantale*

begin

lemma *res-peirce*:

assumes $\forall x y. x^\circ \cdot -(x \cdot y) \leq -y$

shows $((x \cdot y) \sqcap z^\circ = \perp) = ((y \cdot z) \sqcap x^\circ = \perp)$

proof

assume $(x \cdot y) \sqcap z^\circ = \perp$

hence $z^\circ \leq -(x \cdot y)$

by (*simp add: local.inf commute local.inf-shunt*)

thus $(y \cdot z) \sqcap x^\circ = \perp$

by (*metis assms local.inf-shunt local.inv-conjugate local.inv-contrav local.inv-invol local.mult-isol local.order.trans*)

next

assume $(y \cdot z) \sqcap x^\circ = \perp$

hence $x^\circ \leq -(y \cdot z)$

using *local.compl-le-swap1 local.inf-shunt* **by** *blast*

thus $(x \cdot y) \sqcap z^\circ = \perp$

by (*metis assms local.dual-order.trans local.inf-shunt local.inv-conjugate local.inv-contrav local.mult-isol*)

qed

lemma *res-schroeder1*:

assumes $\forall x y. x^\circ \cdot -(x \cdot y) \leq -y$

shows $((x \cdot y) \sqcap z = \perp) = (y \sqcap (x^\circ \cdot z) = \perp)$

proof

assume *h*: $(x \cdot y) \sqcap z = \perp$

hence $z \leq -(x \cdot y)$

by (*simp add: local.inf commute local.inf-shunt*)

thus $y \sqcap (x^\circ \cdot z) = \perp$

by (*metis assms local.dual-order.trans local.inf commute local.inf-shunt local.mult-isol*)

next

assume $y \sqcap (x^\circ \cdot z) = \perp$

hence $y \leq -(x^\circ \cdot z)$

by (*simp add: local.inf-shunt*)

thus $(x \cdot y) \sqcap z = \perp$

by (*metis assms local.inf-shunt local.inv-invol local.order-trans mult-isol*)

qed

lemma *res-schroeder2*:

assumes $\forall x y. x^\circ \cdot -(x \cdot y) \leq -y$

shows $((x \cdot y) \sqcap z = \perp) = (x \sqcap (z \cdot y^\circ) = \perp)$

by (*metis assms local.inv-invol local.res-peirce local.res-schroeder1*)

lemma *res-mod*:

```

assumes  $\forall x y. x^\circ \cdot -(x \cdot y) \leq -y$ 
shows  $(x \cdot y) \sqcap z \leq (x \sqcap (z \cdot y^\circ)) \cdot y$ 
proof –
  have  $(x \cdot y) \sqcap z = ((x \sqcap ((z \cdot y^\circ) \sqcup -(z \cdot y^\circ))) \cdot y) \sqcap z$ 
    by simp
  also have  $\dots = (((x \sqcap (z \cdot y^\circ)) \cdot y) \sqcap z) \sqcup (((x \sqcap -(z \cdot y^\circ)) \cdot y) \sqcap z)$ 
    using local.chaq.wsqq.distrib-left local.inf commute local.sup-distr by presburger
  also have  $\dots \leq ((x \sqcap (z \cdot y^\circ)) \cdot y) \sqcup ((x \cdot y) \sqcap -(z \cdot y^\circ)) \cdot y \sqcap z$ 
    by (metis assms local.inf commute local.inf-compl-bot-right local.sup.orderI local.sup-inf-absorb res-schroeder2)
  also have  $\dots \leq ((x \sqcap (z \cdot y^\circ)) \cdot y) \sqcup ((x \cdot y) \sqcap -z \sqcap z)$ 
    by (metis assms local.dual-order.eq-iff local.inf commute local.inf-compl-bot-right res-schroeder2)
  also have  $\dots \leq ((x \sqcap (z \cdot y^\circ)) \cdot y)$ 
    by (simp add: local.inf commute)
  finally show ?thesis.
qed

end

```

The strong Gelfand property (name by Palmigiano and Re) is important for dioids and Kleene algebras. The modular law is a convenient axiom for relational quantales, in a setting where the underlying lattice is not boolean.

```

class quantale-converse = involutive-quantale +
  assumes strong-gelfand:  $x \leq x \cdot x^\circ \cdot x$ 

```

```

begin

```

```

lemma do-gelfand [simp]:  $do\ x \cdot do\ x \cdot do\ x = do\ x$ 
  apply (rule order.antisym)
  using local.do-subid local.h-seq local.mult-isol apply fastforce
  by (metis local.inv-do local.strong-gelfand)

```

```

lemma cd-gelfand [simp]:  $cd\ x \cdot cd\ x \cdot cd\ x = cd\ x$ 
  by (metis do-gelfand local.do-inv)

```

```

lemma do-idem [simp]:  $do\ x \cdot do\ x = do\ x$ 
  apply (rule order.antisym)
  using local.do-subid local.mult-isol apply fastforce
  by (metis do-gelfand local.do-subid local.eq-refl local.nsrnqo.mult-oner local.psrpq.mult-isol-var)

```

```

lemma cd-idem [simp]:  $cd\ x \cdot cd\ x = cd\ x$ 
  by (metis do-idem local.do-inv)

```

```

lemma dodo [simp]:  $do\ (do\ x) = do\ x$ 

```

```

proof –
  have  $do\ (do\ x) = 1 \sqcap (do\ x \cdot do\ x)$ 
    using local.do-def local.inv-do by force

```

also have $\dots = 1 \sqcap do\ x$
by *simp*
also have $\dots = do\ x$
by (*simp add: local.do-def*)
finally show *?thesis*.
qed

lemma *cdcd* [*simp*]: $cd\ (cd\ x) = cd\ x$
using *cd-idem local.cd-def local.inv-cd* **by** *force*

lemma *docd-compat* [*simp*]: $do\ (cd\ x) = cd\ x$
proof –
have $do\ (cd\ x) = do\ (do\ (x^\circ))$
by (*simp add: local.do-inv*)
also have $\dots = do\ (x^\circ)$
by *simp*
also have $\dots = cd\ x$
by (*simp add: local.do-inv*)
finally show *?thesis*.
qed

lemma *cddo-compat* [*simp*]: $cd\ (do\ x) = do\ x$
by (*metis docd-compat local.cd-inv local.inv-do*)

end

sublocale *quantale-converse* \subseteq *convqka: kleene-algebra-converse* (\sqcup) (\cdot) $1 \perp$ (\leq)
($<$) *invol qstar*
by *unfold-locales (simp add: local.strong-gelfand)*

6.3 Dedekind quantales

class *dedekind-quantale* = *involutive-quantale* +
assumes *modular-law*: $(x \cdot y) \sqcap z \leq (x \sqcap (z \cdot y^\circ)) \cdot y$

begin

sublocale *convdqka: kleene-algebra-converse* (\sqcup) (\cdot) $1 \perp$ (\leq) ($<$) *invol qstar*
by *unfold-locales (metis local.inf.absorb2 local.le-top local.modular-law local.top-greatest)*

subclass *quantale-converse*
by *unfold-locales (simp add: local.convdqka.strong-gelfand)*

lemma *modular-2* [*simp*]: $((x \sqcap (z \cdot y^\circ)) \cdot y) \sqcap z = (x \cdot y) \sqcap z$
apply (*rule order.antisym*)
using *local.inf.cobounded1 local.inf-mono local.mult-isor local.order-refl* **apply**
presburger
by (*simp add: local.modular-law*)

lemma modular-1 [*simp*]: $(x \cdot (y \sqcap (x^\circ \cdot z))) \sqcap z = (x \cdot y) \sqcap z$
by (*metis local.inv-binf local.inv-contrav local.inv-invol modular-2*)

lemma modular3: $(x \cdot y) \sqcap z \leq x \cdot (y \sqcap (x^\circ \cdot z))$
by (*metis local.inf.cobounded1 modular-1*)

The name Dedekind quantale owes to the following formula, which is equivalent to the modular law. Dedekind quantales are called modular quantales in Rosenthal's book on quantaloids (to be precise: he discusses modular quantaloids, but the notion of modular quantale is then obvious).

lemma dedekind: $(x \cdot y) \sqcap z \leq (x \sqcap (z \cdot y^\circ)) \cdot (y \sqcap (x^\circ \cdot z))$

proof–

have $(x \cdot y) \sqcap z = (x \cdot (y \sqcap (x^\circ \cdot z))) \sqcap z$

by *simp*

also have $\dots \leq (x \sqcap (z \cdot (y \sqcap (x^\circ \cdot z))^\circ)) \cdot (y \sqcap (x^\circ \cdot z))$

using *local.modular-law* **by** *presburger*

also have $\dots = (x \sqcap (z \cdot (y^\circ \sqcap (z^\circ \cdot x)))) \cdot (y \sqcap (x^\circ \cdot z))$

by *simp*

also have $\dots \leq (x \sqcap (z \cdot y^\circ)) \cdot (y \sqcap (x^\circ \cdot z))$

using *local.inf.commute modular-1* **by** *fastforce*

finally show *?thesis*.

qed

lemma peirce: $((x \cdot y) \sqcap z^\circ = \perp) = ((y \cdot z) \sqcap x^\circ = \perp)$

proof

assume $(x \cdot y) \sqcap z^\circ = \perp$

hence $((x \cdot y) \sqcap z^\circ) \cdot y^\circ = \perp$

by *simp*

hence $(z^\circ \cdot y^\circ) \sqcap x = \perp$

by (*metis local.inf.commute local.inv-invol local.le-bot local.modular-law*)

hence $((y \cdot z) \sqcap x^\circ)^\circ = \perp^\circ$

by *simp*

thus $(y \cdot z) \sqcap x^\circ = \perp$

by (*metis local.inv-invol*)

next

assume $h: (y \cdot z) \sqcap x^\circ = \perp$

hence $z^\circ \cdot ((y \cdot z) \sqcap x^\circ) = \perp$

by *simp*

hence $(y^\circ \cdot x^\circ) \sqcap z = \perp$

by (*metis h local.inf.commute local.inv-invol local.le-bot local.mult-botr modular3*)

hence $((x \cdot y) \sqcap z^\circ)^\circ = \perp^\circ$

by *simp*

thus $(x \cdot y) \sqcap z^\circ = \perp$

by (*metis local.inv-invol*)

qed

lemma schroeder-1: $((x \cdot y) \sqcap z = \perp) = (y \sqcap (x^\circ \cdot z) = \perp)$

by (*metis local.inf.commute local.inf-bot-right local.inv-invol local.mult-botr mod-*

ular-1)

lemma *schroeder-2*: $((x \cdot y) \sqcap z = \perp) = (x \sqcap (z \cdot y^\circ) = \perp)$
by (*metis local.inv-invol peirce schroeder-1*)

lemma *modular-eq2*: $y \sqcap (z \cdot x^\circ) \leq w \implies (y \cdot x) \sqcap z \leq w \cdot x$
by (*meson local.dual-order.trans local.eq-refl local.h-w1 local.modular-law*)

lemma *lla-top-aux*: $p \leq 1 \implies ((x \leq p \cdot x) = (x \leq p \cdot \top))$

proof

assume *h*: $p \leq 1$

and *h1*: $x \leq p \cdot x$

thus $x \leq p \cdot \top$

by (*meson local.mult-isol local.order-trans local.top-greatest*)

next

assume *h*: $p \leq 1$

and $x \leq p \cdot \top$

hence $x = (p \cdot \top) \sqcap x$

using *local.inf.absorb-iff2* **by** *auto*

also have $\dots \leq p \cdot (\top \sqcap (p^\circ \cdot x))$

using *modular3* **by** *blast*

also have $\dots = p \cdot p \cdot x$

by (*simp add: h local.convdqka.subid-conv mult-assoc*)

finally show $x \leq p \cdot x$

by (*metis h local.dual-order.trans local.mult-isol local.nsrnqo.mult-one1*)

qed

Next we turn to properties of domain and codomain in Dedekind quantales.

lemma *lra-top-aux*: $p \leq 1 \implies ((x \leq x \cdot p) = (x \leq \top \cdot p))$

by (*metis convdqka.subid-conv local.inf.absorb-iff2 local.mult-1-right local.psrpq.subdistl local.sup.absorb-iff2 local.top-greatest modular-eq2*)

lemma *lla*: $p \leq 1 \implies ((do\ x \leq p) = (x \leq p \cdot \top))$

proof

assume *a1*: $x \leq p \cdot \top$

assume *a2*: $p \leq 1$

have *f3*: $x \cdot \top \leq p \cdot \top \cdot \top$

by (*simp add: a1 local.mult-isol*)

have *f4*: $p \cdot do\ x \leq p$

by (*simp add: local.do-subid local.uqka.star-inductr-var-equiv local.uwqlka.star-subid*)

have $x \cdot \top \leq p \cdot \top$

using *f3* **by** (*simp add: local.mult.semigroup-axioms semigroup.assoc*)

thus $do\ x \leq p$

using *f4 a2 lla-top-aux local.do-le-top local.inf.bounded-iff local.order-trans* **by**

blast

next

assume *a1*: $do\ x \leq p$

assume *a2*: $p \leq 1$

hence $do\ x \cdot x \leq p \cdot x$

by (*simp add: a1 local.mult-isor*)
 hence $x \leq p \cdot x$
 using *a1 local.do-def modular-eq2* by *fastforce*
 thus $x \leq p \cdot \top$
 by (*simp add: a2 lla-top-aux*)
qed

lemma *lla-Inf*: $do\ x = \sqcap \{p. x \leq p \cdot \top \wedge p \leq 1\}$
 apply (*rule order.antisym*)
 using *lla local.Inf-greatest* apply *fastforce*
 by (*metis CollectI lla local.Inf-lower local.do-subid local.order.refl*)

lemma *lra*: $p \leq 1 \implies ((cd\ x \leq p) = (x \leq \top \cdot p))$
 by (*metis invqka.inv-adj lla local.convdqka.subid-conv local.do-inv local.inv-contrav local.inv-top*)

lemma *lra-Inf*: $cd\ x = \sqcap \{p. x \leq \top \cdot p \wedge p \leq 1\}$
 apply (*rule order.antisym*)
 using *local.Inf-greatest lra* apply *fastforce*
 by (*metis CollectI local.Inf-lower local.cd-subid local.order.refl lra*)

lemma *lla-var*: $p \leq 1 \implies ((do\ x \leq p) = (x \leq p \cdot x))$
 by (*simp add: lla lla-top-aux*)

lemma *lla-Inf-var*: $do\ x = \sqcap \{p. x \leq p \cdot x \wedge p \leq 1\}$
 apply (*rule order.antisym*)
 using *lla-var local.Inf-greatest* apply *fastforce*
 by (*metis CollectI lla-var local.Inf-lower local.do-subid local.order.refl*)

lemma *lra-var*: $p \leq 1 \implies ((cd\ x \leq p) = (x \leq x \cdot p))$
 by (*simp add: lra lra-top-aux*)

lemma *lra-Inf-var*: $cd\ x = \sqcap \{p. x \leq x \cdot p \wedge p \leq 1\}$
 apply (*rule order.antisym*)
 using *local.Inf-greatest lra-var* apply *fastforce*
 by (*metis CollectI local.Inf-lower local.cd-subid local.order.refl lra-var*)

lemma *do-top*: $do\ x = 1 \sqcap (x \cdot \top)$

proof–

have $1 \sqcap (x \cdot \top) = 1 \sqcap (x \cdot (\top \sqcap x^\circ \cdot 1))$
 by (*metis local.inf commute local.inf-top.left-neutral modular-1*)
 also have $\dots = do\ x$
 by (*simp add: local.do-def*)
 finally show *?thesis..*

qed

lemma *cd-top*: $cd\ x = 1 \sqcap (\top \cdot x)$

by (*metis do-top invqka.inv-one local.do-inv local.inv-binf local.inv-cd local.inv-contrav local.inv-invol local.inv-top*)

We start deriving the axioms of modal semirings and modal quantales.

lemma *do-absorp*: $x \leq do\ x \cdot x$
using *lla-var local.do-subid* **by** *blast*

lemma *cd-absorp*: $x \leq x \cdot cd\ x$
using *local.cd-subid lra-var* **by** *blast*

lemma *do-absorp-eq* [*simp*]: $do\ x \cdot x = x$
using *do-absorp local.do-subid local.dual-order.antisym local.h-w1* **by** *fastforce*

lemma *cd-absorp-eq* [*simp*]: $x \cdot cd\ x = x$
by (*metis do-top local.do-inv local.inf-top.right-neutral local.nsrnq.mult-oner modular-1*)

lemma *do-top2*: $x \cdot \top = do\ x \cdot \top$

proof (*rule order.antisym*)

have $x \cdot \top = do\ x \cdot x \cdot \top$

by *simp*

also have $\dots \leq do\ x \cdot \top \cdot \top$

using *local.psrpq.mult-double-iso local.top-greatest* **by** *presburger*

also have $\dots = do\ x \cdot \top$

by (*simp add: local.mult.semigroup-axioms semigroup.assoc*)

finally show $x \cdot \top \leq do\ x \cdot \top$.

have $do\ x \cdot \top = (1 \sqcap (x \cdot \top)) \cdot \top$

by (*simp add: do-top*)

also have $\dots \leq (1 \cdot \top) \sqcap (x \cdot \top \cdot \top)$

by (*simp add: local.mult-isor*)

also have $\dots = x \cdot \top$

by (*simp add: mult-assoc*)

finally show $do\ x \cdot \top \leq x \cdot \top$.

qed

lemma *cd-top2*: $\top \cdot x = \top \cdot cd\ x$

by (*metis do-top2 local.do-inv local.inv-cd local.inv-contrav local.inv-invol local.inv-top*)

lemma *do-local* [*simp*]: $do\ (x \cdot do\ y) = do\ (x \cdot y)$

proof–

have $do\ (x \cdot do\ y) = 1 \sqcap (x \cdot do\ y \cdot \top)$

using *do-top* **by** *force*

also have $\dots = 1 \sqcap (x \cdot y \cdot \top)$

using *do-top2 mult-assoc* **by** *force*

also have $\dots = do\ (x \cdot y)$

by (*simp add: do-top*)

finally show *?thesis*

by *force*

qed

lemma *cd-local* [*simp*]: $cd\ (cd\ x \cdot y) = cd\ (x \cdot y)$

by (*metis cd-top cd-top2 mult-assoc*)

lemma *do-fix-subid*: $(do\ x = x) = (x \leq 1)$

proof

assume $do\ x = x$

thus $x \leq 1$

by (*metis local.do-subid*)

next

assume $a: x \leq 1$

hence $x = do\ x \cdot x$

by *simp*

hence $b: x \leq do\ x$

by (*metis a local.mult-isol local.nsrngo.mult-oner*)

have $x \cdot x \leq x$

using *a local.mult-isol* by *fastforce*

hence $do\ x \leq x$

by (*simp add: a lla-var local.le-top lra-top-aux*)

thus $do\ x = x$

by (*simp add: b local.dual-order.antisym*)

qed

lemma *cd-fix-subid*: $(cd\ x = x) = (x \leq 1)$

by (*metis local.convdqka.subid-conv local.do-inv local.do-fix-subid local.docd-compat*)

lemma *do-inf2*: $do\ (do\ x \sqcap do\ y) = do\ x \sqcap do\ y$

using *do-top local.do-fix-subid local.inf.assoc* by *auto*

lemma *do-inf-comp*: $do\ x \cdot do\ y = do\ x \sqcap do\ y$

by (*smt (verit, best) local.do-def local.do-idem local.do-fix-subid local.dodo local.dual-order.trans local.inf-commute local.inf-idem local.inv-contrav local.inv-do local.mult-1-right local.mult-isol modular-1 mult-assoc*)

lemma *cd-inf-comp*: $cd\ x \cdot cd\ y = cd\ x \sqcap cd\ y$

by (*metis do-inf-comp local.docd-compat*)

lemma *subid-mult-meet*: $p \leq 1 \implies q \leq 1 \implies p \cdot q = p \sqcap q$

by (*metis do-inf-comp local.do-fix-subid*)

lemma *dodo-sup*: $do\ (do\ x \sqcup do\ y) = do\ x \sqcup do\ y$

proof–

have $do\ (do\ x \sqcup do\ y) = 1 \sqcap ((do\ x \sqcup do\ y) \cdot (do\ x \sqcup do\ y)^\circ)$

using *local.do-def* by *blast*

also have $\dots = 1 \sqcap ((do\ x \sqcup do\ y) \cdot (do\ x \sqcup do\ y))$

by *simp*

also have $\dots = 1 \sqcap (do\ x \sqcup do\ y)$

using *local.do-subid local.dodo local.inf.idem local.le-supI subid-mult-meet* by

presburger

also have $\dots = do\ x \sqcup do\ y$

by (*simp add: local.do-def local.inf-absorb2*)

finally show *?thesis*.
qed

lemma *do-sup [simp]*: $do (x \sqcup y) = do x \sqcup do y$

proof –

have $do (x \sqcup y) = 1 \sqcap ((x \sqcup y) \cdot \top)$

by (*simp add: do-top*)

also have $\dots = 1 \sqcap (x \cdot \top \sqcup y \cdot \top)$

by *simp*

also have $\dots = 1 \sqcap (do x \cdot \top \sqcup do y \cdot \top)$

using *do-top2* **by** *force*

also have $\dots = 1 \sqcap ((do x \sqcup do y) \cdot \top)$

by *simp*

also have $\dots = do (do x \sqcup do y)$

by (*simp add: do-top*)

finally show *?thesis*

by (*simp add: dodo-sup*)

qed

lemma *cdcd-sup*: $cd (cd x \sqcup cd y) = cd x \sqcup cd y$

using *local.cd-fix-subid* **by** *fastforce*

lemma *cd-sup [simp]*: $cd (x \sqcup y) = cd x \sqcup cd y$

by (*metis do-sup local.do-inv local.inv-binsup*)

Next we show that Dedekind quantales are modal quantales, hence also modal semirings.

sublocale *dmq*: *dc-modal-quantale* 1 (\cdot) *Inf Sup* (\sqcap) (\leq) ($<$) (\sqcup) \perp \top *cd do*

proof

show $\bigwedge x. cd x \leq 1$

by (*simp add: cd-top*)

show $\bigwedge x. do x \leq 1$

by (*simp add: do-top*)

qed *simp-all*

lemma *do-top3 [simp]*: $do (x \cdot \top) = do x$

using *dmq.coddual.le-top dmq.dqmsr.domain-1'' local.do-iso local.order.antisym*

by *presburger*

lemma *cd-top3 [simp]*: $cd (\top \cdot x) = cd x$

by (*simp add: cd-top dmq.coddual.mult-assoc*)

lemma *dodo-Sup-pres*: $do (\bigsqcup x \in X. do x) = (\bigsqcup x \in X. do x)$

by (*simp add: local.SUP-least local.do-fix-subid*)

The domain elements form a complete Heyting algebra.

lemma *do-complete-heyting*: $do x \sqcap (\bigsqcup y \in Y. do y) = (\bigsqcup y \in Y. do x \sqcap do y)$

proof –

have $do x \sqcap (\bigsqcup y \in Y. do y) = do x \cdot (\bigsqcup y \in Y. do y)$

by (*metis do-inf-comp dodo-Sup-pres*)
 also have $\dots = (\bigsqcup y \in Y. do\ x \cdot do\ y)$
 by (*simp add: dmq.coddual.Sup-distr image-image*)
 also have $\dots = (\bigsqcup y \in Y. do\ x \sqcap do\ y)$
 using *do-inf-comp* by *force*
 finally show *?thesis*.
 qed

lemma *cdcd-Sup-pres*: $cd\ (\bigsqcup x \in X. cd\ x) = (\bigsqcup x \in X. cd\ x)$
 by (*simp add: local.SUP-least local.cd-fix-subid*)

lemma *cd-complete-heyting*: $cd\ x \sqcap (\bigsqcup y \in Y. cd\ y) = (\bigsqcup y \in Y. cd\ x \sqcap cd\ y)$
proof –

have $cd\ x \sqcap (\bigsqcup y \in Y. cd\ y) = cd\ x \cdot (\bigsqcup y \in Y. cd\ y)$
 by (*metis cd-inf-comp cdcd-Sup-pres*)
 also have $\dots = (\bigsqcup y \in Y. cd\ x \cdot cd\ y)$
 by (*simp add: dmq.coddual.Sup-distr image-image*)
 also have $\dots = (\bigsqcup y \in Y. cd\ x \sqcap cd\ y)$
 using *cd-inf-comp* by *force*
 finally show *?thesis*.
 qed

lemma *subid-complete-heyting*:

assumes $p \leq 1$
 and $\forall q \in Q. q \leq 1$
 shows $p \sqcap (\bigsqcup Q) = (\bigsqcup q \in Q. p \sqcap q)$

proof –

have $a: do\ p = p$
 by (*simp add: asms(1) local.do-fix-subid*)
 have $b: \forall q \in Q. do\ q = q$
 using *asms(2) local.do-fix-subid* by *presburger*
 hence $p \sqcap (\bigsqcup Q) = do\ p \sqcap (\bigsqcup q \in Q. do\ q)$
 by (*simp add: a*)
 also have $\dots = (\bigsqcup q \in Q. do\ p \sqcap do\ q)$
 using *do-complete-heyting* by *blast*
 also have $\dots = (\bigsqcup q \in Q. p \sqcap q)$
 using *a b* by *force*
 finally show *?thesis*.
 qed

Next we show that domain and codomain preserve arbitrary Sups.

lemma *do-Sup-pres-aux*: $(\bigsqcup x \in X. do\ x \cdot \top) = (\bigsqcup x \in X. do\ x \cdot \top)$
 by (*smt (verit, best) Sup.SUP-cong image-image*)

lemma *do-Sup-pres*: $do\ (\bigsqcup x \in X. x) = (\bigsqcup x \in X. do\ x)$

proof –

have $do\ (\bigsqcup x \in X. x) = 1 \sqcap ((\bigsqcup x \in X. x) \cdot \top)$
 by (*simp add: do-top*)
 also have $\dots = 1 \sqcap (\bigsqcup x \in X. x \cdot \top)$

by (*simp add: dmq.coddual.Sup-distl*)
 also have $\dots = 1 \sqcap (\bigsqcup x \in X. do\ x \cdot \top)$
 using *do-top2* by *force*
 also have $\dots = 1 \sqcap (\bigsqcup x \in do\ 'X. x \cdot \top)$
 using *do-Sup-pres-aux* by *presburger*
 also have $\dots = 1 \sqcap ((\bigsqcup x \in X. do\ x) \cdot \top)$
 using *dmq.coddual.Sup-distl* by *presburger*
 also have $\dots = do\ (\bigsqcup x \in X. do\ x)$
 by (*simp add: do-top*)
 finally show *?thesis*
 using *dodo-Sup-pres* by *presburger*
 qed

lemma *cd-Sup-pres*: $cd\ (\bigsqcup x \in X. x) = (\bigsqcup x \in X. cd\ x)$

proof –

have $cd\ (\bigsqcup x \in X. x) = do\ ((\bigsqcup x \in X. x)^\circ)$
 using *local.do-inv* by *presburger*
 also have $\dots = do\ (\bigsqcup x \in X. x^\circ)$
 by *simp*
 also have $\dots = (\bigsqcup x \in X. do\ (x^\circ))$
 by (*metis do-Sup-pres image-ident image-image*)
 also have $\dots = (\bigsqcup x \in X. cd\ x)$
 using *local.do-inv* by *presburger*
 finally show *?thesis*.

qed

lemma *do-inf*: $do\ (x \sqcap y) = 1 \sqcap (y \cdot x^\circ)$

by (*smt (z3) do-absorp-eq invqka.inv-one local.do-def local.inf-commute local.inv-binf local.inv-contrav local.inv-invol local.mult-1-right modular-1 modular-2 mult-assoc*)

lemma *cd-inf*: $cd\ (x \sqcap y) = 1 \sqcap (y^\circ \cdot x)$

by (*metis do-inf local.do-inv local.inv-binf local.inv-invol*)

lemma *do-bres-prop*: $p \leq 1 \implies do\ (x \rightarrow p \cdot \top) = 1 \sqcap (x \rightarrow p \cdot \top)$

by (*simp add: do-top*)

lemma *cd-fres-prop*: $p \leq 1 \implies cd\ (\top \cdot p \leftarrow x) = 1 \sqcap (\top \cdot p \leftarrow x)$

by (*simp add: cd-top*)

lemma *do-meet-prop*: $(do\ p \cdot x) \sqcap (x \cdot do\ q) = do\ p \cdot x \cdot do\ q$

proof (*rule order.antisym*)

have $x \sqcap (do\ p \cdot x \cdot do\ q) \leq do\ p \cdot x$

by (*simp add: dmq.dqmsr.dom-subid-aux2'' local.inf.coboundedI2*)

thus $(do\ p \cdot x) \sqcap (x \cdot do\ q) \leq do\ p \cdot x \cdot do\ q$

by (*simp add: local.inf.commute modular-eq2*)

next

have $do\ p \cdot x \cdot do\ q = (do\ p \cdot x \cdot do\ q) \sqcap (do\ p \cdot x \cdot do\ q)$

by *simp*

also have $\dots \leq (do\ p \cdot x) \sqcap (x \cdot do\ q)$

using *dmq.dqmsr.dom-subid-aux2 dmq.dqmsr.dom-subid-aux2'' local.psrpq.mult-isol-var*
by *auto*
finally show $do\ p \cdot x \cdot do\ q \leq (do\ p \cdot x) \sqcap (x \cdot do\ q)$.
qed

lemma *subid-meet-prop*: $p \leq 1 \implies q \leq 1 \implies (p \cdot x) \sqcap (x \cdot q) = p \cdot x \cdot q$
by (*metis do-fix-subid do-meet-prop*)

Next we consider box and diamond operators like in modal semirings and modal quantales. These are inherited from domain quantales. Diamonds are defined with respect to domain and codomain. The box operators are defined as Sups and hence right adjoints of diamonds.

abbreviation *do-dia* \equiv *dmq.fd'*

abbreviation *cd-dia* \equiv *dmq.bd'*

abbreviation *do-box* \equiv *dmq.bb*

abbreviation *cd-box* \equiv *dmq.fb*

In the sense of modal logic, the domain-based diamond is a backward operator, the codomain-based one a forward operator. These are related by opposition/converse.

lemma *do-dia-cd-dia-conv*: $p \leq 1 \implies do\text{-}dia\ (x^\circ)\ p = cd\text{-}dia\ x\ p$
by (*simp add: convdqka.subid-conv dmq.coddual.dqmsr.fd-def dmq.dqmsr.fd-def local.cd-def local.do-def*)

lemma *cd-dia-do-dia-conv*: $p \leq 1 \implies cd\text{-}dia\ (x^\circ)\ p = do\text{-}dia\ x\ p$
by (*metis do-dia-cd-dia-conv local.inv-invol*)

Diamonds preserve sups in both arguments.

lemma *do-dia-Sup*: $do\text{-}dia\ (\bigsqcup X)\ p = (\bigsqcup x \in X.\ do\text{-}dia\ x\ p)$

proof –

have $do\text{-}dia\ (\bigsqcup X)\ p = do\ ((\bigsqcup X) \cdot p)$
by (*simp add: dmq.dqmsr.fd-def*)
also have $\dots = do\ (\bigsqcup x \in X.\ x \cdot p)$
using *local.Sup-distr* **by** *fastforce*
also have $\dots = (\bigsqcup x \in X.\ do\ (x \cdot p))$
by (*metis do-Sup-pres image-ident image-image*)
also have $\dots = (\bigsqcup x \in X.\ do\text{-}dia\ x\ p)$
using *dmq.dqmsr.fd-def* **by** *fastforce*
finally show *?thesis*.

qed

lemma *do-dia-Sup2*: $do\text{-}dia\ x\ (\bigsqcup P) = (\bigsqcup p \in P.\ do\text{-}dia\ x\ p)$

proof –

have $do\text{-}dia\ x\ (\bigsqcup P) = do\ (x \cdot (\bigsqcup P))$
by (*simp add: dmq.dqmsr.fd-def*)

also have $\dots = (\bigsqcup p \in P. \text{do } (x \cdot p))$
by (*metis dmq.coddual.Sup-distr do-Sup-pres image-ident image-image*)
also have $\dots = (\bigsqcup p \in P. \text{do-dia } x \ p)$
using *dmq.dqmsr.fd-def* **by** *auto*
finally show *?thesis*.
qed

lemma *cd-dia-Sup*: $\text{cd-dia } (\bigsqcup X) \ p = (\bigsqcup x \in X. \text{cd-dia } x \ p)$

proof –
have $\text{cd-dia } (\bigsqcup X) \ p = \text{cd } (p \cdot (\bigsqcup X))$
by (*simp add: dmq.coddual.dqmsr.fd-def*)
also have $\dots = \text{cd } (\bigsqcup x \in X. p \cdot x)$
using *dmq.coddual.Sup-distr* **by** *auto*
also have $\dots = (\bigsqcup x \in X. \text{cd } (p \cdot x))$
by (*metis cd-Sup-pres image-ident image-image*)
also have $\dots = (\bigsqcup x \in X. \text{cd-dia } x \ p)$
using *dmq.coddual.dqmsr.fd-def* **by** *force*
finally show *?thesis*.
qed

lemma *cd-dia-Sup2*: $\text{cd-dia } x \ (\bigsqcup P) = (\bigsqcup p \in P. \text{cd-dia } x \ p)$

proof –
have $\text{cd-dia } x \ (\bigsqcup P) = \text{cd } ((\bigsqcup P) \cdot x)$
by (*simp add: dmq.coddual.dqmsr.fd-def*)
also have $\dots = (\bigsqcup p \in P. \text{cd } (p \cdot x))$
by (*metis cd-Sup-pres dmq.coddual.Sup-distl image-ident image-image*)
also have $\dots = (\bigsqcup p \in P. \text{cd-dia } x \ p)$
using *dmq.coddual.dqmsr.fd-def* **by** *auto*
finally show *?thesis*.
qed

The domain-based box is a forward operator, the codomain-based on a backward one. These interact again with respect to converse.

lemma *do-box-var*: $p \leq 1 \implies \text{do-box } x \ p = \bigsqcup \{q. \text{do-dia } x \ q \leq p \wedge q \leq 1\}$
by (*smt (verit, best) Collect-cong dmq.bb-def dmq.dqmsr.fdia-d-simp local.do-fix-subid local.dodo*)

lemma *cd-box-var*: $p \leq 1 \implies \text{cd-box } x \ p = \bigsqcup \{q. \text{cd-dia } x \ q \leq p \wedge q \leq 1\}$
by (*smt (verit, best) Collect-cong dmq.coddual.dqmsr.fdia-d-simp dmq.fb-def local.cd-fix-subid local.cdcd*)

lemma *do-box-cd-box-conv*: $p \leq 1 \implies \text{do-box } (x^\circ) \ p = \text{cd-box } x \ p$

proof –
assume $a: p \leq 1$
hence $\text{do-box } (x^\circ) \ p = \bigsqcup \{q. \text{do-dia } (x^\circ) \ q \leq p \wedge q \leq 1\}$
by (*simp add: do-box-var*)
also have $\dots = \bigsqcup \{q. \text{cd-dia } x \ q \leq p \wedge q \leq 1\}$
by (*metis do-dia-cd-dia-conv*)
also have $\dots = \text{cd-box } x \ p$

using a *cd-box-var* by *auto*
 finally show *?thesis*.
 qed

lemma *cd-box-do-box-conv*: $p \leq 1 \implies \text{cd-box } (x^\circ) p = \text{do-box } x p$
 by (*metis do-box-cd-box-conv local.inv-invol*)

lemma *do-box-subid*: $\text{do-box } x p \leq 1$
 using *dmq.bb-def local.Sup-le-iff* by *force*

lemma *cd-box-subid*: $p \leq 1 \implies \text{cd-box } x p \leq 1$
 by (*metis do-box-subid local.do-box-cd-box-conv*)

Next we prove that boxes and diamonds are adjoints, and then demodalisation laws known from modal semirings.

lemma *do-dia-do-box-galois*:

assumes $p \leq 1$
 and $q \leq 1$
 shows $(\text{do-dia } x p \leq q) = (p \leq \text{do-box } x q)$

proof

show $\text{do-dia } x p \leq q \implies p \leq \text{do-box } x q$
 by (*simp add: assms do-box-var local.Sup-upper*)

next

assume $p \leq \text{do-box } x q$
 hence $\text{do-dia } x p \leq \text{do } (x \cdot \bigsqcup \{t. \text{do-dia } x t \leq q \wedge t \leq 1\})$
 by (*simp add: assms(2) local.dmq.dqmsr.fd-def local.do-box-var local.do-iso local.mult-isol*)

also have $\dots = \bigsqcup \{ \text{do } (x \cdot t) \mid t. \text{do-dia } x t \leq q \wedge t \leq 1 \}$

by (*metis do-Sup-pres image-ident image-image local.Sup-distl setcompr-eq-image*)

also have $\dots = \bigsqcup \{ \text{do-dia } x t \mid t. \text{do-dia } x t \leq q \wedge t \leq 1 \}$

using *local.dmq.dqmsr.fd-def* by *fastforce*

also have $\dots \leq q$

using *local.Sup-le-iff* by *blast*

finally have $\text{do-dia } x p \leq q$.

thus $p \leq \text{do-box } x q \implies \text{do-dia } x p \leq q$.

qed

lemma *cd-dia-cd-box-galois*:

assumes $p \leq 1$

and $q \leq 1$

shows $(\text{cd-dia } x p \leq q) = (p \leq \text{cd-box } x q)$

by (*metis assms do-box-cd-box-conv do-dia-cd-dia-conv do-dia-do-box-galois*)

lemma *do-dia-demod-subid*:

assumes $p \leq 1$

and $q \leq 1$

shows $(\text{do-dia } x p \leq q) = (x \cdot p \leq q \cdot x)$

by (*metis assms dmq.dqmsr.fdemodalisation2 local.do-fix-subid*)

The demodalisation laws have variants based on residuals.

lemma *do-dia-demod-subid-fres*:

assumes $p \leq 1$

and $q \leq 1$

shows $(do\text{-}dia\ x\ p \leq q) = (p \leq x \rightarrow q \cdot x)$

by (*simp add: assms do-dia-demod-subid local.bres-galois*)

lemma *do-dia-demod-subid-var*:

assumes $p \leq 1$

and $q \leq 1$

shows $(do\text{-}dia\ x\ p \leq q) = (x \cdot p \leq q \cdot \top)$

by (*simp add: assms(2) dmq.dqmsr.fd-def lla*)

lemma *do-dia-demod-subid-var-fres*:

assumes $p \leq 1$

and $q \leq 1$

shows $(do\text{-}dia\ x\ p \leq q) = (p \leq x \rightarrow q \cdot \top)$

by (*simp add: assms do-dia-demod-subid-var local.bres-galois*)

lemma *cd-dia-demod-subid*:

assumes $p \leq 1$

and $q \leq 1$

shows $(cd\text{-}dia\ x\ p \leq q) = (p \cdot x \leq x \cdot q)$

by (*metis assms dmq.coddual.dqmsr.fdemodalisation2 local.cd-fix-subid*)

lemma *cd-dia-demod-subid-fres*:

assumes $p \leq 1$

and $q \leq 1$

shows $(cd\text{-}dia\ x\ p \leq q) = (p \leq x \cdot q \leftarrow x)$

by (*simp add: assms cd-dia-demod-subid local.fres-galois*)

lemma *cd-dia-demod-subid-var*:

assumes $p \leq 1$

and $q \leq 1$

shows $(cd\text{-}dia\ x\ p \leq q) = (p \cdot x \leq \top \cdot q)$

by (*simp add: assms(2) dmq.coddual.dqmsr.fd-def tra*)

lemma *cd-dia-demod-subid-var-fres*:

assumes $p \leq 1$

and $q \leq 1$

shows $(cd\text{-}dia\ x\ p \leq q) = (p \leq \top \cdot q \leftarrow x)$

by (*simp add: assms cd-dia-demod-subid-var local.fres-galois*)

lemma *do-box-iso*:

assumes $p \leq 1$

and $q \leq 1$

and $p \leq q$

shows $do\text{-}box\ x\ p \leq do\text{-}box\ x\ q$

by (*meson assms do-box-subid do-dia-do-box-galois local.order.refl local.order.trans*)

lemma *cd-box-iso*:
assumes $p \leq 1$
and $q \leq 1$
and $p \leq q$
shows $cd\text{-box } x \ p \leq cd\text{-box } x \ q$
by (*metis assms do-box-cd-box-conv do-box-iso*)

lemma *do-box-demod-subid*:
assumes $p \leq 1$
and $q \leq 1$
shows $(p \leq do\text{-box } x \ q) = (x \cdot p \leq q \cdot x)$
using *assms do-dia-do-box-galois local.do-dia-demod-subid* **by force**

lemma *do-box-demod-subid-bres*:
assumes $p \leq 1$
and $q \leq 1$
shows $(p \leq do\text{-box } x \ q) = (p \leq x \rightarrow q \cdot x)$
by (*simp add: assms do-box-demod-subid local.bres-galois*)

lemma *do-box-demod-subid-var*:
assumes $p \leq 1$
and $q \leq 1$
shows $(p \leq do\text{-box } x \ q) = (x \cdot p \leq q \cdot \top)$
using *assms do-dia-demod-subid-var do-dia-do-box-galois* **by auto**

lemma *do-box-demod-subid-var-bres*:
assumes $p \leq 1$
and $q \leq 1$
shows $(p \leq do\text{-box } x \ q) = (p \leq x \rightarrow q \cdot \top)$
by (*simp add: assms do-box-demod-subid-var local.bres-galois*)

lemma *do-box-demod-subid-var-bres-do*:
assumes $p \leq 1$
and $q \leq 1$
shows $(p \leq do\text{-box } x \ q) = (p \leq do (x \rightarrow q \cdot \top))$
by (*simp add: assms do-box-demod-subid-var-bres do-top*)

lemma *cd-box-demod-subid*:
assumes $p \leq 1$
and $q \leq 1$
shows $(p \leq cd\text{-box } x \ q) = (p \cdot x \leq x \cdot q)$
using *assms local.cd-dia-cd-box-galois local.cd-dia-demod-subid* **by force**

lemma *cd-box-demod-subid-fres*:
assumes $p \leq 1$
and $q \leq 1$
shows $(p \leq cd\text{-box } x \ q) = (p \leq x \cdot q \leftarrow x)$
by (*simp add: assms cd-box-demod-subid local.fres-galois*)

lemma *cd-box-demod-subid-var*:
assumes $p \leq 1$
and $q \leq 1$
shows $(p \leq \text{cd-box } x \ q) = (p \cdot x \leq \top \cdot q)$
using *assms cd-dia-cd-box-galois cd-dia-demod-subid-var* **by force**

lemma *cd-box-demod-subid-var-fres*:
assumes $p \leq 1$
and $q \leq 1$
shows $(p \leq \text{cd-box } x \ q) = (p \leq \top \cdot q \leftarrow x)$
by (*simp add: assms cd-box-demod-subid-var local.fres-galois*)

We substitute demodalisation inequalities for diamonds in the definitions of boxes.

lemma *do-box-var2*: $p \leq 1 \implies \text{do-box } x \ p = \bigsqcup \{q. x \cdot q \leq p \cdot x \wedge q \leq 1\}$
unfolding *do-box-var* **by** (*meson do-dia-demod-subid*)

lemma *do-box-bres1*: $p \leq 1 \implies \text{do-box } x \ p = \bigsqcup \{q. q \leq x \rightarrow p \cdot x \wedge q \leq 1\}$
unfolding *do-box-var* **by** (*meson do-dia-demod-subid-fres*)

lemma *do-box-bres2*: $p \leq 1 \implies \text{do-box } x \ p = \bigsqcup \{q. q \leq x \rightarrow p \cdot \top \wedge q \leq 1\}$
unfolding *do-box-var* **by** (*simp add: dmq.dqmsr.fd-def lla local.bres-galois*)

lemma *do-box-var3*: $p \leq 1 \implies \text{do-box } x \ p = \bigsqcup \{q. x \cdot q \leq p \cdot \top \wedge q \leq 1\}$
unfolding *do-box-var* **using** *dmq.dqmsr.fd-def lla* **by force**

lemma *cd-box-var2*: $p \leq 1 \implies \text{cd-box } x \ p = \bigsqcup \{q. q \cdot x \leq x \cdot p \wedge q \leq 1\}$
unfolding *cd-box-var* **by** (*metis cd-dia-demod-subid*)

lemma *cd-box-var3*: $p \leq 1 \implies \text{cd-box } x \ p = \bigsqcup \{q. q \cdot x \leq \top \cdot p \wedge q \leq 1\}$
unfolding *cd-box-var* **by** (*simp add: dmq.coddual.dqmsr.fd-def lra*)

Using these results we get a simple characterisation of boxes via domain and codomain. Similar laws can be found implicitly in Doornbos, Backhouse and van der Woude's paper on a calculational approach to mathematical induction, which speaks about wlp operators instead modal operators.

lemma *bres-do-box*: $p \leq 1 \implies \text{do-box } x \ p = \text{do } (x \rightarrow p \cdot \top)$
by (*meson do-box-demod-subid-var-bres-do do-box-subid local.cd-fix-subid local.cddo-compat local.dual-order.antisym local.eq-refl*)

lemma *bres-do-box-var*: $p \leq 1 \implies \text{do-box } x \ p = 1 \sqcap (x \rightarrow p \cdot \top)$
using *bres-do-box do-bres-prop* **by force**

lemma *bres-do-box-top*: $p \leq 1 \implies (\text{do-box } x \ p) \cdot \top = x \rightarrow p \cdot \top$
using *bres-do-box do-top2* **by auto**

lemma *fres-cd-box*: $p \leq 1 \implies \text{cd-box } x \ p = \text{cd } (\top \cdot p \leftarrow x)$
proof –

```

assume  $h0: p \leq 1$ 
{fix  $q$ 
assume  $h1: q \leq 1$ 
have  $(q \leq \text{cd-box } x \ p) = (q \cdot x \leq \top \cdot p)$ 
  by (simp add: cd-box-demod-subid-var h0 h1)
also have  $\dots = (q \leq \top \cdot p \leftarrow x)$ 
  by (simp add: local.fres-galois)
also have  $\dots = (q \leq 1 \sqcap (\top \cdot p \leftarrow x))$ 
  by (simp add: h1)
also have  $\dots = (q \leq \text{cd } (\top \cdot p \leftarrow x))$ 
  by (simp add: cd-fres-prop h0)
finally have  $(q \leq \text{cd-box } x \ p) = (q \leq \text{cd } (\top \cdot p \leftarrow x)).\}$ 
hence  $\forall y. y \leq \text{cd-box } x \ p \longleftrightarrow y \leq \text{cd } (\top \cdot p \leftarrow x)$ 
  by (meson cd-box-subid dmq.coddual.dqmsr.dpd3 h0 local.dual-order.trans)
thus ?thesis
using local.dual-order.antisym by blast
qed

```

```

lemma fres-cd-box-var:  $p \leq 1 \implies \text{cd-box } x \ p = 1 \sqcap (\top \cdot p \leftarrow x)$ 
  by (simp add: cd-fres-prop fres-cd-box)

```

```

lemma fres-cd-box-top:  $p \leq 1 \implies \top \cdot \text{cd-box } x \ p = \top \cdot p \leftarrow x$ 
  using cd-top2 fres-cd-box by auto

```

Next we show that the box operators act on the complete Heyting algebra of subidentities.

```

lemma cd-box-act:
  assumes  $p \leq 1$ 
  shows  $\text{cd-box } (x \cdot y) \ p = \text{cd-box } x \ (\text{cd-box } y \ p)$ 
proof –
  {fix  $q$ 
    assume  $a: q \leq 1$ 
    hence  $(q \leq \text{cd-box } (x \cdot y) \ p) = (\text{cd-dia } (x \cdot y) \ q \leq p)$ 
      by (simp add: assms cd-dia-cd-box-galois)
    also have  $\dots = (\text{cd-dia } y \ (\text{cd-dia } x \ q) \leq p)$ 
      by (simp add: local.dmq.coddual.dqmsr.fdia-mult)
    also have  $\dots = (\text{cd-dia } x \ q \leq \text{cd-box } y \ p)$ 
      using assms cd-dia-cd-box-galois cd-top dmq.coddual.dqmsr.fd-def by force
    also have  $\dots = (q \leq \text{cd-box } x \ (\text{cd-box } y \ p))$ 
      by (simp add: a assms cd-dia-cd-box-galois local.cd-box-subid)
    finally have  $(q \leq \text{cd-box } (x \cdot y) \ p) = (q \leq \text{cd-box } x \ (\text{cd-box } y \ p)).\}$ 
  thus ?thesis
  by (meson assms local.cd-box-subid local.dual-order.eq-iff)
qed

```

```

lemma do-box-act:
  assumes  $p \leq 1$ 
  shows  $\text{do-box } (x \cdot y) \ p = \text{do-box } y \ (\text{do-box } x \ p)$ 
  by (smt (verit) assms cd-box-act local.cd-box-do-box-conv local.do-box-subid lo-)

```

cal.inv-contrav)

Next we show that the box operators are Sup reversing in the first and Inf preserving in the second argument.

lemma *do-box-sup-inf*: $p \leq 1 \implies \text{do-box } (x \sqcup y) p = \text{do-box } x p \cdot \text{do-box } y p$

proof –

assume *h1*: $p \leq 1$

{fix *q*

assume *h2*: $q \leq 1$

hence $(q \leq \text{do-box } (x \sqcup y) p) = (\text{do-dia } (x \sqcup y) q \leq p)$

by (*simp add: do-dia-do-box-galois h1*)

also have $\dots = (\text{do-dia } x q \leq p \wedge \text{do-dia } y q \leq p)$

by (*simp add: dmq.dqmsr.fdia-add2*)

also have $\dots = (q \leq \text{do-box } x p \wedge q \leq \text{do-box } y p)$

by (*simp add: do-dia-do-box-galois h1 h2*)

also have $\dots = (q \leq \text{do-box } x p \cdot \text{do-box } y p)$

by (*simp add: do-box-subid subid-mult-meet*)

finally have $(q \leq \text{do-box } (x \sqcup y) p) = (q \leq \text{do-box } x p \cdot \text{do-box } y p).$

hence $\forall z. z \leq \text{do-box } (x \sqcup y) p \iff z \leq \text{do-box } x p \cdot \text{do-box } y p$

by (*metis do-box-subid local.dual-order.trans local.inf.boundedE subid-mult-meet*)

thus *?thesis*

using *local.dual-order.antisym* **by** *blast*

qed

lemma *do-box-sup-inf-var*: $p \leq 1 \implies \text{do-box } (x \sqcup y) p = \text{do-box } x p \sqcap \text{do-box } y p$

by (*simp add: do-box-subid do-box-sup-inf subid-mult-meet*)

lemma *do-box-Sup-Inf*:

assumes $X \neq \{\}$

and $p \leq 1$

shows $\text{do-box } (\bigsqcup X) p = (\bigsqcap x \in X. \text{do-box } x p)$

proof –

{fix *q*

assume *h*: $q \leq 1$

hence $(q \leq \text{do-box } (\bigsqcup X) p) = (\text{do-dia } (\bigsqcup X) q \leq p)$

by (*simp add: do-dia-do-box-galois assms(2)*)

also have $\dots = ((\bigsqcup x \in X. \text{do-dia } x q) \leq p)$

using *do-dia-Sup* **by** *force*

also have $\dots = (\forall x \in X. \text{do-dia } x q \leq p)$

by (*simp add: local.SUP-le-iff*)

also have $\dots = (\forall x \in X. q \leq \text{do-box } x p)$

using *do-dia-do-box-galois assms(2) h* **by** *blast*

also have $\dots = (q \leq (\bigsqcap x \in X. \text{do-box } x p))$

by (*simp add: local.le-INF-iff*)

finally have $(q \leq \text{do-box } (\bigsqcup X) p) = (q \leq (\bigsqcap x \in X. \text{do-box } x p)).$

hence $\forall y. y \leq \text{do-box } (\bigsqcup X) p \iff y \leq (\bigsqcap x \in X. \text{do-box } x p)$

by (*smt (verit, ccfv-threshold) assms(1) do-box-subid local.INF-le-SUP local.SUP-least local.order-trans*)

thus *?thesis*

using *local.dual-order.antisym* by *blast*
qed

lemma *do-box-Sup-Inf2*:

assumes $P \neq \{\}$

and $\forall p \in P. p \leq 1$

shows $do\text{-}box\ x (\prod P) = (\prod p \in P. do\text{-}box\ x\ p)$

proof –

have $h0: (\prod p \in P. do\text{-}box\ x\ p) \leq 1$

by (*meson all-not-in-conv assms(1) do-box-subid local.INF-lower2*)

{**fix** q

assume $h1: q \leq 1$

hence $(q \leq do\text{-}box\ x (\prod P)) = (do\text{-}dia\ x\ q \leq \prod P)$

 by (*simp add: assms do-dia-do-box-galois local.Inf-less-eq*)

also have $\dots = (\forall p \in P. do\text{-}dia\ x\ q \leq p)$

 using *local.le-Inf-iff* by *blast*

also have $\dots = (\forall p \in P. q \leq do\text{-}box\ x\ p)$

 using *assms(2) do-dia-do-box-galois h1* by *auto*

also have $\dots = (q \leq (\prod p \in P. do\text{-}box\ x\ p))$

 by (*simp add: local.le-INF-iff*)

finally have $(q \leq do\text{-}box\ x (\prod P)) = (q \leq (\prod p \in P. do\text{-}box\ x\ p)).$ }

thus *?thesis*

 using *do-box-subid h0 local.dual-order.antisym* by *blast*

qed

lemma *cd-box-sup-inf*: $p \leq 1 \implies cd\text{-}box\ (x \sqcup y)\ p = cd\text{-}box\ x\ p \cdot cd\text{-}box\ y\ p$

by (*metis do-box-cd-box-conv do-box-sup-inf local.inv-binsup*)

lemma *cd-box-sup-inf-var*: $p \leq 1 \implies cd\text{-}box\ (x \sqcup y)\ p = cd\text{-}box\ x\ p \sqcap cd\text{-}box\ y\ p$

by (*simp add: cd-box-subid cd-box-sup-inf subid-mult-meet*)

lemma *cd-box-Sup-Inf*:

assumes $X \neq \{\}$

and $p \leq 1$

shows $cd\text{-}box\ (\bigsqcup X)\ p = (\prod x \in X. cd\text{-}box\ x\ p)$

proof –

have $cd\text{-}box\ (\bigsqcup X)\ p = do\text{-}box\ ((\bigsqcup X)^\circ)\ p$

 using *assms(2) do-box-cd-box-conv* by *presburger*

also have $\dots = (\prod y \in \{x^\circ \mid x. x \in X\}. do\text{-}box\ y\ p)$

 by (*simp add: Setcompr-eq-image assms do-box-Sup-Inf*)

also have $\dots = (\prod x \in X. do\text{-}box\ (x^\circ)\ p)$

 by (*simp add: Setcompr-eq-image image-image*)

also have $\dots = (\prod x \in X. cd\text{-}box\ x\ p)$

 using *assms(2) do-box-cd-box-conv* by *force*

finally show *?thesis*.

qed

lemma *cd-box-Sup-Inf2*:

assumes $P \neq \{\}$

and $\forall p \in P. p \leq 1$
shows $cd\text{-box } x (\bigsqcap P) = (\bigsqcap p \in P. cd\text{-box } x p)$
proof –
have $cd\text{-box } x (\bigsqcap P) = do\text{-box } (x^\circ) (\bigsqcap P)$
by (*simp add: assms do-box-cd-box-conv local.Inf-less-eq*)
also have $\dots = (\bigsqcap p \in P. do\text{-box } (x^\circ) p)$
using *assms do-box-Sup-Inf2 by presburger*
also have $\dots = (\bigsqcap p \in P. cd\text{-box } x p)$
using *assms(2) do-box-cd-box-conv by force*
finally show *?thesis.*
qed

Next we define an antidomain operation in the style of modal semirings. A natural condition is that the antidomain of an element is the greatest test that cannot be left-composed with that elements, and hence a greatest left annihilator. The definition of anticodomain is similar. As we are not in a boolean domain algebra, we cannot expect that the antidomain of the antidomain yields the domain or that the union of a domain element with the corresponding antidomain element equals one.

definition $ado\ x = \bigsqcup \{p. p \cdot x = \perp \wedge p \leq 1\}$

definition $acd\ x = \bigsqcup \{p. x \cdot p = \perp \wedge p \leq 1\}$

lemma *ado-acd*: $ado\ (x^\circ) = acd\ x$

unfolding *ado-def acd-def*

by (*metis convdqka.subid-conv invqka.inv-zero local.inv-contrav local.inv-invol*)

lemma *acd-ado*: $acd\ (x^\circ) = ado\ x$

unfolding *ado-def acd-def*

by (*metis acd-def ado-acd ado-def local.inv-invol*)

lemma *ado-left-zero* [*simp*]: $ado\ x \cdot x = \perp$

unfolding *ado-def*

using *dmq.coddual.Sup-distl by auto*

lemma *acd-right-zero* [*simp*]: $x \cdot acd\ x = \perp$

unfolding *acd-def*

by (*simp add: dmq.coddual.h-Sup local.dual-order.antisym*)

lemma *ado-greatest*: $p \leq 1 \implies p \cdot x = \perp \implies p \leq ado\ x$

by (*simp add: ado-def local.Sup-upper*)

lemma *acd-greatest*: $p \leq 1 \implies x \cdot p = \perp \implies p \leq acd\ x$

by (*simp add: acd-def local.Sup-upper*)

lemma *ado-subid*: $ado\ x \leq 1$

using *ado-def local.Sup-le-iff by force*

lemma *acd-subid*: $acd\ x \leq 1$

by (*simp add: acd-def local.Sup-le-iff*)

lemma *ado-left-zero-iff*: $p \leq 1 \implies (p \leq ado\ x) = (p \cdot x = \perp)$

by (*metis ado-greatest ado-left-zero local.bot-unique local.mult-isor*)

lemma *acd-right-zero-iff*: $p \leq 1 \implies (p \leq acd\ x) = (x \cdot p = \perp)$

by (*metis acd-greatest acd-right-zero local.bot-unique local.mult-isol*)

This gives an equational characterisation of antidomain and anticodomain.

lemma *ado-cd-bot*: $ado\ x = cd\ (\perp \leftarrow x)$

proof –

{fix p

assume $h: p \leq 1$

hence $(p \leq ado\ x) = (p \cdot x = \perp)$

by (*simp add: ado-left-zero-iff*)

also have $\dots = (p \leq \perp \leftarrow x)$

using *local.bot-unique local.fres-galois* **by** *blast*

also have $\dots = (p \leq 1 \sqcap (\perp \leftarrow x))$

by (*simp add: h*)

also have $\dots = (p \leq cd\ (\perp \leftarrow x))$

by (*metis cd-fres-prop local.bot-least local.mult-botr*)

finally have $(p \leq ado\ x) = (p \leq cd\ (\perp \leftarrow x)).$

hence $\forall y. y \leq ado\ x \iff y \leq cd\ (\perp \leftarrow x)$

using *ado-subid local.cd-subid local.dual-order.trans* **by** *blast*

thus *?thesis*

using *local.dual-order.antisym* **by** *blast*

qed

lemma *acd-do-bot*: $acd\ x = do\ (x \rightarrow \perp)$

by (*metis ado-acd ado-cd-bot invqka.inv-zero local.bres-fres-conv local.cd-inv local.inv-invol*)

lemma *ado-cd-bot-id*: $ado\ x = 1 \sqcap (\perp \leftarrow x)$

by (*metis ado-cd-bot cd-fres-prop local.cd-bot local.cd-subid local.mult-botr*)

lemma *acd-do-bot-id*: $acd\ x = 1 \sqcap (x \rightarrow \perp)$

by (*metis acd-do-bot do-bres-prop local.bot-least local.mult-botl*)

lemma *ado-cd-bot-var*: $ado\ x = cd\ (\perp \leftarrow do\ x)$

by (*metis ado-cd-bot do-top2 local.fres-bot-top local.fres-curry*)

lemma *acd-do-bot-var*: $acd\ x = do\ (cd\ x \rightarrow \perp)$

by (*metis acd-do-bot cd-top2 local.bres-curry local.bres-top-bot*)

lemma *ado-do-bot*: $ado\ x = do\ (do\ x \rightarrow \perp)$

using *acd-ado acd-do-bot-var local.cd-inv* **by** *auto*

lemma $do\ x = ado\ (ado\ x)$

oops

lemma *acd-cd-bot*: $acd\ x = cd\ (\perp \leftarrow cd\ x)$
by (*metis ado-acd ado-cd-bot-var local.cd-inv local.inv-invol*)

lemma *ado-do-bot-var*: $ado\ x = 1 \sqcap (do\ x \rightarrow \perp)$
by (*metis ado-do-bot do-bres-prop local.bot-unique local.bres-bot-bot local.bres-canc1 local.do-bot local.do-subid*)

lemma *acd-cd-bot-var*: $acd\ x = 1 \sqcap (\perp \leftarrow cd\ x)$
by (*metis acd-cd-bot acd-right-zero cd-top local.fres-top-top2*)

Domain and codomain are compatible with the boxes.

lemma *cd-box-ado*: $cd\ box\ x \perp = ado\ x$
by (*simp add: ado-cd-bot fres-cd-box*)

lemma *do-box-acd*: $do\ box\ x \perp = acd\ x$
by (*simp add: acd-do-bot bres-do-box*)

lemma *ado-subid-prop*: $p \leq 1 \implies ado\ p = 1 \sqcap (p \rightarrow \perp)$
by (*metis ado-do-bot-var do-fix-subid*)

lemma *ado-do*: $p \leq 1 \implies ado\ p = do\ (p \rightarrow \perp)$
using *ado-do-bot do-fix-subid* **by** *force*

lemma *ado-do-compl*: $ado\ x \cdot do\ x = \perp$
using *dmq.dqmsr.dom-weakly-local* **by** *force*

lemma $ado\ x \sqcup do\ x = \top$
oops

lemma $\forall x\ p.\ \exists f.\ 1 \sqcap (\top \cdot p \leftarrow x) = 1 \sqcap (\perp \leftarrow (x \rightarrow p \cdot \top))$
oops

lemma $cd\ box\ x\ p = ado\ (x \cdot ado\ p)$
oops

lemma *ad-do-bot [simp]*: $(1 \sqcap (do\ x \rightarrow \perp)) \cdot do\ x = \perp$
using *ado-do-bot-var ado-left-zero dmq.dqmsr.dom-weakly-local* **by** *presburger*

lemma *do-heyting-galois*: $(do\ x \cdot do\ y \leq do\ z) = (do\ x \leq 1 \sqcap (do\ y \rightarrow do\ z))$
by (*metis dmq.dqmsr.dsg4 dmq.mqdual.cod-subid local.bres-galois local.le-inf-iff*)

lemma *do-heyting-galois-var*: $(do\ x \cdot do\ y \leq do\ z) = (do\ x \leq cd\ box\ (do\ y)\ (do\ z))$
by (*metis cd-dia-cd-box-galois cd-fix-subid dmq.coddual.dqmsr.fd-def dmq.dqmsr.dom-mult-closed local.do-subid*)

Antidomain is therefore Heyting negation.

lemma *ado-heyting-negation*: $ado (do x) = cd\text{-}box (do x) \perp$
by (*simp add: cd-box-ado*)

lemma *ad-ax1* [*simp*]: $(1 \sqcap (do x \rightarrow \perp)) \cdot x = \perp$
by (*simp add: local.dmq.dqmsr.dom-weakly-local*)

lemma $1 \sqcap (do (1 \sqcap (do x \rightarrow \perp)) \rightarrow \perp) = do x$
oops

lemma $p \leq 1 \implies do\text{-}dia x p = 1 \sqcap (cd\text{-}box x (1 \sqcap (p \rightarrow \perp)) \rightarrow \perp)$
oops

lemma $p \leq 1 \implies cd\text{-}box x p = 1 \sqcap (do\text{-}dia x (1 \sqcap (p \rightarrow \perp)) \rightarrow \perp)$
oops

lemma $p \leq 1 \implies cd\text{-}dia x p = 1 \sqcap (do\text{-}box x (1 \sqcap (p \rightarrow \perp)) \rightarrow \perp)$
oops

lemma $p \leq 1 \implies do\text{-}box x p = 1 \sqcap (cd\text{-}dia x (1 \sqcap (p \rightarrow \perp)) \rightarrow \perp)$
oops

end

6.4 Boolean Dedekind quantales

class *distributive-dedekind-quantale* = *distrib-unital-quantale* + *dedekind-quantale*

class *boolean-dedekind-quantale* = *bool-unital-quantale* + *distributive-dedekind-quantale*

begin

lemma *ad-do-bot* [*simp*]: $(1 - do x) \cdot do x = \perp$
by (*simp add: local.diff-eq local.inf-shunt local.subid-mult-meet*)

lemma *ad-ax1* [*simp*]: $(1 - do x) \cdot x = \perp$
by (*simp add: local.dmq.dqmsr.dom-weakly-local*)

lemma *ad-do* [*simp*]: $1 - do (1 - do x) = do x$

proof –

have $1 - do (1 - do x) = 1 - (1 - do x)$

by (*metis local.diff-eq local.do-fix-subid local.inf.cobounded1*)

also have $\dots = 1 \sqcap -(1 \sqcap - do x)$

by (*simp add: local.diff-eq*)

also have $\dots = do x$

by (*simp add: local.chaq.wswq.distrib-left local.do-top*)

finally show *?thesis*.

qed

lemma *ad-ax2*: $1 - do (x \cdot y) \sqcup (1 - do (x \cdot (1 - do (1 - do y)))) = 1 - do$

$(x \cdot (1 - do (1 - do y)))$
by *simp*

lemma *ad-ax3* [*simp*]: $do\ x \sqcup (1 - do\ x) = 1$

proof –

have $do\ x \sqcup (1 - do\ x) = do\ x \sqcup (1 \sqcap -do\ x)$

using *local.diff-eq* **by** *force*

also have $\dots = 1 \sqcap (do\ x \sqcup -do\ x)$

by (*simp add: local.chaq.wsq.distrib-left local.do-top*)

also have $\dots = 1$

by *simp*

finally show *?thesis*.

qed

sublocale *bdad*: *antidomain-semiring* $\lambda x. 1 - do\ x \sqcup (\cdot) 1 \perp - -$
by *unfold-locales simp-all*

sublocale *bdadka*: *antidomain-kleene-algebra* $\lambda x. 1 - do\ x \sqcup (\cdot) 1 \perp - - qstar..$

sublocale *bdar*: *antirange-semiring* $\sqcup (\cdot) 1 \perp \lambda x. 1 - cd\ x - -$
apply *unfold-locales*

apply (*metis ad-ax1 ad-do dmq.mqs.local-var local.docd-compat*)

apply (*metis ad-do local.cddo-compat local.dmq.coddual.dqmsr.dsr2 local.docd-compat local.sup.idem*)

by (*metis bdad.a-subid' bdad.as3 local.convdqka.subid-conv local.do-inv*)

sublocale *bdaka*: *antirange-kleene-algebra* $\sqcup (\cdot) 1 \perp - - qstar\ \lambda x. 1 - cd\ x..$

sublocale *bmod*: *modal-semiring-simp* $\lambda x. 1 - do\ x \sqcup (\cdot) 1 \perp - - \lambda x. 1 - cd\ x..$

sublocale *bmod*: *modal-kleene-algebra-simp* $\sqcup (\cdot) 1 \perp - - qstar\ \lambda x. 1 - do\ x\ \lambda x. 1 - cd\ x..$

lemma *inv-neg*: $(-x)^\circ = -(x^\circ)$

by (*metis local.diff-eq local.diff-shunt-var local.dual-order.eq-iff local.inf commute local.inv-conjugate local.inv-galois*)

lemma *residuation*: $x^\circ \cdot -(x \cdot y) \leq -y$

by (*metis local.inf commute local.inf-compl-bot local.inf-shunt local.schroeder-1*)

lemma *bres-prop*: $x \rightarrow y = -(x^\circ \cdot -y)$

by (*metis local.ba-dual.dual-iff local.bres-canc1 local.bres-galois-imp local.compl-le-swap1 local.dmq.coddual.h-w1 local.dual-order.antisym local.inv-invol residuation*)

lemma *fres-prop*: $x \leftarrow y = -(x \cdot y^\circ)$

using *bres-prop inv-neg local.fres-bres-conv* **by** *auto*

lemma *do-dia-fdia*: $do\ dia\ x\ p = bdad.fdia\ x\ p$

by (*simp add: local.bdad.fdia-def local.dmq.dqmsr.fd-def*)

lemma *cd-dia-bdia*: $cd\text{-}dia\ x\ p = bdar.bdia\ x\ p$
by (*metis ad-do bdar.ardual.a-subid' bdar.bdia-def local.cd-fix-subid local.dmq.coddual.dqmsr.fd-def local.docd-compat*)

lemma *do-dia-fbox-de-morgan*: $p \leq 1 \implies do\text{-}dia\ x\ p = 1 - bdad.fbox\ x\ (1 - p)$
by (*smt (verit, ccfv-SIG) ad-do bdad.a-closure bdad.am-d-def bdad.fbox-def local.dmq.dqmsr.fd-def local.do-fix-subid*)

lemma *fbox-do-dia-de-morgan*: $p \leq 1 \implies bdad.fbox\ x\ p = 1 - do\text{-}dia\ x\ (1 - p)$
using *bdad.fbox-def local.dmq.dqmsr.fd-def local.do-fix-subid* **by** *force*

lemma *cd-dia-bbox-de-morgan*: $p \leq 1 \implies cd\text{-}dia\ x\ p = 1 - bdar.bbox\ x\ (1 - p)$
by (*smt (verit, best) ad-do bdar.bbox-def bdar.bdia-def cd-dia-bdia local.cd-fix-subid local.do-subid local.docd-compat*)

lemma *bbox-cd-dia-de-morgan*: $p \leq 1 \implies bdar.bbox\ x\ p = 1 - cd\text{-}dia\ x\ (1 - p)$
using *bdar.bbox-def local.cd-fix-subid local.dmq.coddual.dqmsr.fd-def* **by** *force*

lemma *do-box-bbox*: $p \leq 1 \implies do\text{-}box\ x\ p = bdar.bbox\ x\ p$

proof –

assume *a*: $p \leq 1$

{fix *q*

assume *b*: $q \leq 1$

hence $(q \leq do\text{-}box\ x\ p) = (x \cdot q \leq p \cdot x)$

by (*simp add: a local.do-box-demod-subid*)

also have $\dots = (x \cdot cd\ q \leq cd\ p \cdot x)$

by (*metis a b local.cd-fix-subid*)

also have $\dots = (cd\ q \leq bdar.bbox\ x\ p)$

by (*metis bdar.bbox-def bdar.bdia-def cd-dia-bdia local.bdar.ardual.a-closure' local.bdar.ardual.ans-d-def local.bdar.ardual.dnsz.dsg1 local.bdar.ardual.fbox-demodalisation3 local.bdar.ardual.fbox-one local.dmq.coddual.dqmsr.fd-def local.nsrnqo.mult-oner*)

also have $\dots = (q \leq bdar.bbox\ x\ p)$

using *b local.cd-fix-subid* **by** *force*

finally have $(q \leq do\text{-}box\ x\ p) = (q \leq bdar.bbox\ x\ p).$

thus *?thesis*

by (*metis bdar.bbox-def local.bdar.ardual.a-subid' local.do-box-subid local.dual-order.antisym local.eq-refl*)

qed

lemma *cd-box-fbox*: $p \leq 1 \implies cd\text{-}box\ x\ p = bdad.fbox\ x\ p$

by (*smt (verit, ccfv-SIG) bdar.bbox-def do-box-bbox local.bdad.fbox-def local.cd-dia-do-dia-conv local.cd-inv local.cd-fix-subid local.cd-top local.diff-eq local.dmq.bb-def local.dmq.coddual.dqmsr.fd-def local.dmq.coddual.mult-assoc local.dmq.dqmsr.fd-def local.dmq.fb-def local.do-box-cd-box-conv local.do-fix-subid local.do-top local.inf.cobounded1*)

lemma *do-dia-cd-box-de-morgan*: $p \leq 1 \implies do\text{-}dia\ x\ p = 1 - cd\text{-}box\ x\ (1 - p)$

by (*simp add: cd-box-fbox local.diff-eq local.do-dia-fbox-de-morgan*)

lemma *cd-box-do-dia-de-morgan*: $p \leq 1 \implies \text{cd-box } x \ p = 1 - \text{do-dia } x \ (1 - p)$
by (*simp add: cd-box-fbox local.fbox-do-dia-de-morgan*)

lemma *cd-dia-do-box-de-morgan*: $p \leq 1 \implies \text{cd-dia } x \ p = 1 - \text{do-box } x \ (1 - p)$
by (*simp add: do-box-bbox local.cd-dia-bbox-de-morgan local.diff-eq*)

lemma *do-box-cd-dia-de-morgan*: $p \leq 1 \implies \text{do-box } x \ p = 1 - \text{cd-dia } x \ (1 - p)$
by (*simp add: do-box-bbox local.bbox-cd-dia-de-morgan*)

end

class *dc-involutive-modal-quantale* = *dc-modal-quantale* + *involutive-quantale*

begin

sublocale *invmqmka*: *involutive-dr-modal-kleene-algebra* (\sqcup) (\cdot) $1 \perp (\leq) (<)$ *qstar invol dom cod..*

lemma *do-approx-dom*: $\text{do } x \leq \text{dom } x$

proof –

have $\text{do } x = \text{dom } (\text{do } x) \cdot \text{do } x$

by *simp*

also have $\dots \leq \text{dom } (1 \sqcap (x \cdot x^\circ))$

by (*simp add: local.do-def local.dqmsr.domain-subid*)

also have $\dots \leq \text{dom } 1 \sqcap \text{dom } (x \cdot x^\circ)$

using *local.dom-meet-sub* **by** *presburger*

also have $\dots \leq \text{dom } (x \cdot \text{dom } (x^\circ))$

by *simp*

also have $\dots \leq \text{dom } x$

by (*simp add: local.dqmsr.domain-1''*)

finally show *?thesis*.

qed

end

class *dc-modal-quantale-converse* = *dc-involutive-modal-quantale* + *quantale-converse*

sublocale *dc-modal-quantale-converse* \subseteq *invmqmka*: *dr-modal-kleene-algebra-converse* (\sqcup) (\cdot) $1 \perp (\leq) (<)$ *qstar invol dom cod..*

class *dc-modal-quantale-strong-converse* = *dc-involutive-modal-quantale* +

assumes *weak-dom-def*: $\text{dom } x \leq x \cdot x^\circ$

and *weak-cod-def*: $\text{cod } x \leq x^\circ \cdot x$

begin

sublocale *invmqmka*: *dr-modal-kleene-algebra-strong-converse* (\sqcup) (\cdot) $1 \perp (\leq) (<)$
qstar invol dom cod

by (*unfold-locales, simp-all add: local.weak-dom-def local.weak-cod-def*)

```

lemma dom-def: dom x = 1  $\sqcap$  (x · xo)
  using local.do-approx-dom local.do-def local.dual-order.eq-iff local.weak-dom-def
by force

lemma cod-def: cod x = 1  $\sqcap$  (xo · x)
  using local.dom-def local.invmqka.d-conv-cod by auto

lemma do-dom: do x = dom x
  by (simp add: local.do-def local.dom-def)

lemma cd-cod: cd x = cod x
  by (simp add: local.cd-def local.cod-def)

end

class dc-modal-dedekind-quantale = dc-involutive-modal-quantale + dedekind-quantale

class cd-distributive-modal-dedekind-quantale = dc-modal-dedekind-quantale + distrib-unital-quantale

class dc-boolean-modal-dedekind-quantale = dc-modal-dedekind-quantale + bool-unital-quantale

begin

lemma subid-idem: p ≤ 1  $\implies$  p · p = p
  by (simp add: local.subid-mult-meet)

lemma subid-comm: p ≤ 1  $\implies$  q ≤ 1  $\implies$  p · q = q · p
  using local.inf commute local.subid-mult-meet by force

lemma subid-meet-comp: p ≤ 1  $\implies$  q ≤ 1  $\implies$  p  $\sqcap$  q = p · q
  by (simp add: local.subid-mult-meet)

lemma subid-dom: p ≤ 1  $\implies$  dom p = p
proof –
  assume h: p ≤ 1
  have a: p  $\sqcup$  (1  $\sqcap$  -p) = 1
    by (metis h local.inf-sup-absorb local.sup commute local.sup.orderE local.sup-compl-top local.sup-inf-distrib1)
  have b: (1  $\sqcap$  -p)  $\sqcap$  p =  $\perp$ 
    by (simp add: local.inf commute)
  hence dom p = (p  $\sqcup$  (1  $\sqcap$  -p)) · dom p
    by (simp add: a)
  also have ... = p · dom p  $\sqcup$  dom ((1  $\sqcap$  -p) · dom p) · (1  $\sqcap$  -p) · dom p
    using local.dqmsr.dsg1 local.wswq.distrib-right mult-assoc by presburger
  also have ... ≤ p  $\sqcup$  dom ((1  $\sqcap$  -p) · dom p)
    by (metis b h local.dom-subid local.dom-zero local.inf.cobounded1 local.mqdual.cod-local local.mult-botr local.sup.mono subid-comm subid-meet-comp)

```


also have $\dots = p \sqcup \text{dom} ((1 \sqcap -p) \cdot p)$
by *simp*
also have $\dots = p \sqcup \text{dom} \perp$
using *b h subid-meet-comp* **by** *fastforce*
also have $\dots = p$
by *simp*
finally have $\text{dom } p \leq p$.
thus *?thesis*
using *h local.dqmsr.domain-subid local.dual-order.antisym* **by** *presburger*
qed

lemma *do-prop*: $(\text{do } x \leq \text{do } y) = (x \leq \text{do } y \cdot \top)$
by *(simp add: local.lla)*

lemma *do-lla*: $(\text{do } x \leq \text{do } y) = (x \leq \text{do } y \cdot x)$
by *(simp add: local.lla-var)*

lemma *lla-subid*: $p \leq 1 \implies ((\text{dom } x \leq p) = (x \leq p \cdot x))$
by *(metis local.dqmsr.dom-llp subid-dom)*

lemma *dom-do*: $\text{dom } x = \text{do } x$

proof–

have $x \leq \text{do } x \cdot x$
by *simp*
hence $\text{dom } x \leq \text{do } x$
using *lla-subid local.do-subid local.dodo* **by** *presburger*
thus *?thesis*
by *(simp add: local.antisym-conv local.do-approx-dom)*

qed

end

end

References

- [1] A. Armstrong, G. Struth, and T. Weber. Kleene algebra. *Archive of Formal Proofs*, 2013.
- [2] C. Calk, E. Goubault, P. Malbos, and G. Struth. Algebraic coherent confluence and higher globular Kleene algebras. *Logical Methods in Computer Science*, 18(4), 2022.
- [3] C. Calk, P. Malbos, D. Pous, and G. Struth. Higher catoids, higher quantales and their correspondences. *arXiv*, 2307.09253, 2023.
- [4] U. Fahrenberg, C. Johansen, G. Struth, and K. Ziemiański. Catoids and modal convolution algebras. *Algebra Universalis*, 84:10, 2023.

- [5] V. B. F. Gomes, W. Guttmann, P. Höfner, G. Struth, and T. Weber. Kleene algebras with domain. *Archive of Formal Proofs*, 2016.
- [6] V. B. F. Gomes and G. Struth. Modal Kleene algebra applied to program correctness. In *FM 2016*, volume 9995 of *LNCS*, pages 310–325, 2016.
- [7] D. Pous and G. Struth. Dedekind quantaloids as intuitionistic modal algebras. In preparation, 2023.
- [8] P. Resende. Open maps of involutive quantales. *Applied Categorical Structures*, 26:631–644, 2018.
- [9] K. I. Rosenthal. *The Theory of Quantaloids*. Addison Wesley Longman Limited, 1996.
- [10] G. Struth. Quantales. *Archive of Formal Proofs*, 2018.
- [11] A. Tarski. On the calculus of relations. *Journal of Symbolic Logic*, 6(3):73–89, 1941.